


Visual Basic 2005

Windows-Programmierung mit Visual Basic 2005,
Visual Studio 2005 und dem .NET-Framework 2.0

PETER MONADJEMI



Markt+Technik

KOMPENDIUM

Einführung | Arbeitsbuch | Nachschlagewerk

9 Windows Forms für etwas Fortgeschrittene

Es gibt den Maler, der aus der Sonne einen gelben Fleck macht, aber es gibt auch den, der mit Überlegung und Geschick aus einem gelben Fleck eine Sonne macht.

Pablo Picasso (Maler, Grafiker, Bildhauer und visionärer Gestalter von Oberflächen)

Dieses Kapitel richtet sich an jene Leser, die mit dem Prinzip, nach dem ein Button auf einem Formular angeordnet und die Eingabe einer TextBox ausgewertet wird, hinlänglich vertraut sind und nun etwas mehr machen und die restlichen Steuerelemente der Toolbox kennen lernen möchten. Kein Problem, das Windows Forms-Modell hat noch einiges zu bieten. In diesem Kapitel geht es um die etwas fortgeschritteneren Themen, wobei »fortgeschritten« keinesfalls kompliziert bedeutet.

Einer der großen Vorteile des Windows Forms-Modells gegenüber dem alten Modell von VB6 ist nicht, dass dieses einfach nur mehr Features bietet, sondern dass die einzelnen Bestandteile sehr fein aufeinander abgestimmt sind. Da zum Beispiel eine zentrale *Control*-Klasse existiert, von der sich alle Steuerelemente ableiten, verfügen alle Steuerelemente über ein einheitliches Grundgerüst an Eigenschaften und Methoden. Aufgrund der Tatsache, dass eine solche Basisklasse vorhanden ist, können Entwickler ihre eigenen Steuerelemente von dieser Klasse ableiten und diese dadurch mit einem umfangreichen Satz an Grundfunktionalitäten ausstatten. Dieses Prinzip einer modernen Arbeitsteilung findet man an vielen Stellen des Windows Forms-Modells, etwa bei den »Extendern«, die, nachdem sie in ein Formular eingefügt wurden, alle vorhandenen Steuerelemente mit zusätzlichen Eigenschaften ausstatten. Dies spiegelt sich auch in dem Umstand wider, dass eine Toolbar keine Blackbox mehr wie früher darstellt, sondern mit regulären Controls arbeitet und damit sehr flexibel und vor allem auf vertraute Weise erweiterbar ist. In diesem Kapitel geht es ganz konkret um die etwas fortgeschritteneren Steuerelemente, wie Menüs, Toolbars oder TreeViews, mit denen sich anspruchsvollere Benutzeroberflächen realisieren lassen.

Die Themen für dieses Kapitel:

- ➔ Menüs
- ➔ Symbolleisten
- ➔ Speziellere Steuerelemente
- ➔ Ziehen-und-Ablegen-Operationen
- ➔ Layout ist mehr als nur das Anordnen von Steuerelementen

- ➔ Visual Styles bei Windows XP und das Windows-Anwendungsframework
- ➔ Es gibt sie doch – die Steuerelementfelder
- ➔ MDI-Fenster
- ➔ Ein Ersatz für das fehlende Chart-Control

9.1 Menüs


Menüs sind fester Bestandteil praktisch aller modernen Benutzeroberflächen, wengleich sie oft nur eine Alibifunktion haben, da die wirklich wichtigen Funktionen einfacher über Tastaturshortcuts aufgerufen oder in der Symbolleiste angeboten werden, sodass sich kaum jemand die Mühe macht, dafür ein Menü zu öffnen. Menüs sind bei Visual Basic 2005 schnell zusammengestellt, und sie sehen dabei deutlich attraktiver aus als in der Vergangenheit. Den guten, alten Menüeditor von VB6 gibt es schon lange nicht mehr, Menüs werden direkt im Formular zusammengebaut und in einem komfortablen Elementauflistungseditor nachbearbeitet. Dieser Editor wird z.B. über das Kontextmenü des Controls geöffnet.



Neu bei 2.0:

Mit .NET 2.0 wurden die Steuerelemente `MenuStrip` und `ContextMenuStrip` eingeführt, die ein wenig mehr Möglichkeiten und eine etwas modernere Optik bieten. Die folgenden Erklärungen beziehen sich auf diese neuen Controls – die Steuerelemente `MainMenu` und `ContextMenu` aus .NET 1.1 stehen aber natürlich nach wie vor zur Verfügung.


Abbildung 9.1:
Die Hierarchie
der neuen
Strip-Controls



Die Anatomie eines Menüs

Hinter einer Menüleiste steht ein *MenuStrip*-Control, das Sie wie jedes andere Steuerelement auf das Formular ziehen. Auch wenn sich Menüs komplett per Quellcode aufbauen lassen, ist die Umsetzung im Entwurfsmodus sehr praktisch, da sich die Menüstruktur einfach »herunterschreiben« lässt. Hinter jedem Eintrag der Menüleiste steht ein *ToolStripMenuItem*-Objekt. Ein HIER EINGEBEN deutet an, dass sich ein Eintrag der gewünschten Beschriftung machen lässt. Bezüglich der Benennung haben sich die Entwickler bei Visual Basic 2005 eine nette Kleinigkeit einfallen lassen, indem einem Menüeintrag die gewählte Überschrift vorausgeht. Trägt ein Eintrag beispielsweise die Beschriftung »Datei«, lautet der dazugehörige Name »DateiToolStripMenuItem« usw. Lassen Sie sich nicht durch die verschiedenen HIER EINGEBEN-Einträge irritieren, diese verschwinden wieder, wenn Sie hier nichts eingeben.

Abbildung 9.2:
Die Elemente eines
MenuStrip-Menüs



Benötigen Sie ein Menü mit den typischen Standardeinträgen, sollten Sie auf den Eintrag *STANDARDELEMENTE EINFÜGEN* im Eigenschaftenfenster des *MenuStrip*-Controls klicken und sich angenehm überraschen lassen.



Die MenuItem-Klasse

Hinter jedem Menüeintrag steht ein Objekt der *MenuItem*-Klasse, deren wichtigste Mitglieder in Tabelle 9.1 zusammengestellt sind. Ein *MenuItem*-Objekt kann sowohl in ein *MenuStrip*- bzw. *MainMenu*-Control als auch in ein *ContextMenuStrip*- bzw. *ContextMenu*-Control »eingehängt« werden.

Abbildung 9.3:

Menüs sind schnell zusammengestellt und wirken optisch auf der Höhe der Zeit.




Tabelle 9.1: Interessante Mitglieder der MenuItem-Klasse

Eigenschaft	Bedeutung
<i>Alignment</i>	Legt die Ausrichtung des Eintrags (<i>links</i> oder <i>rechts</i>) fest.
<i>AutoToolTip</i>	<i>True</i> , wenn der Text der <i>ToolTipText</i> -Eigenschaft angezeigt werden soll.
<i>Checked</i>	<i>True/False</i> -Wert, über den sich festlegen lässt, ob der Eintrag mit einem Häkchen versehen wird.
<i>OnClick</i>	<i>True</i> , wenn ein mit einem Klick aktiviertes Element diesen Zustand mit einem Häkchen anzeigen soll.
<i>DefaultItem</i>	<i>True/False</i> -Wert, der festlegt, ob der Menüeintrag das Standardmenüelement ist, oder die aktuelle Einstellung zurückgibt.
<i>Enabled</i>	<i>True/False</i> -Wert, der festlegt, ob der Eintrag ausgewählt werden kann, oder die aktuelle Einstellung zurückgibt.
<i>Index</i>	Gibt die Reihenfolge des Eintrags innerhalb des Menüs zurück oder legt diese fest.
<i>IsParent</i>	<i>True/False</i> -Wert, der angibt, ob der Menüeintrag weitere Untereinträge besitzt.
<i>MdiList</i>	<i>True/False</i> -Wert, über den sich festlegen lässt, ob dieser Menüeintrag in einem MDI-Hauptfenster die Liste der Kindfenster anzeigt.
<i>MdiListItem</i>	Gibt das <i>MenuItem</i> -Objekt zurück, das in einem MDI-Hauptfenster die Liste mit den MDI-Kindfenstern anzeigt.
<i>MenuItems</i>	Steht für die Auflistung der <i>MenuItem</i> -Objekte, falls es sich um ein Untermenü handelt.
<i>Mnemonic</i>	Gibt das Zeichen zurück, das auf das &-Zeichen folgt, oder »0«, falls kein Zeichen festgelegt wurde.

Eigenschaft	Bedeutung
<i>OwnerDraw</i>	<i>True-False</i> -Wert, über den eingestellt oder abgefragt wird, ob sich der Menüeintrag selbst zeichnet. In diesem Fall wird ein <i>DrawItem</i> -Ereignis ausgelöst.
<i>RadioCheck</i>	<i>True-False</i> -Wert, über den sich festlegen lässt, ob der Eintrag mit einem Radiobutton anstelle eines Häkchens versehen wird.
<i>Shortcut</i>	Legt die Tastenkombination fest, über die der Menüeintrag aktiviert werden kann, oder gibt die aktuelle Einstellung zurück.
<i>ShowShortcut</i>	<i>True-False</i> -Wert, über den sich festlegen fest, ob der Shortcut angezeigt wird oder nicht.
<i>Text</i>	Legt den Text fest, der in dem Eintrag angezeigt wird, oder gibt diesen zurück.

Gegenüber der eher schlichten *MenuItem*-Klasse von .NET 1.1 wurde die *ToolStripMenuItem*-Klasse von .NET 2.0 stark erweitert. Die wichtigste Neuerung ist natürlich die *Image*-Eigenschaft, mit der es (endlich) möglich wird, jedem Menüeintrag direkt eine Bitmap zuzuweisen. Auch die *Font*-Eigenschaft ist sehr nett, denn damit kann ein Menü (endlich) die zur Auswahl stehenden Schriftarten so anbieten, dass jeder Eintrag in der entsprechenden Schriftart angezeigt wird¹.




Abbildung 9.4: Die ToolStripMenuItem-Klasse von .NET 2.0 wurde gegenüber der MenuItem-Klasse von .NET 1.1 stark erweitert.

¹ Dies ist eine jener Kleinigkeiten, die mir beim Apple Macintosh im Jahre 1985 aufgefallen sind.

Menüereignisse

Die Zeiten sind zum Glück vorbei, in denen ein Menüeintrag lediglich einen harmlosen *Click*-Event lieferte und zum Beispiel das Bewegen des Mauszeigers über einen Eintrag nur per »Subclassing« festgestellt werden konnte. Die *ToolStripMenuItem*-Klasse unterstützt insgesamt 25 Events, wobei neben *Click* und *DoubleClick* die Events *DropDownOpening*, *DropDownOpened* und *MouseHover* erwähnenswert sind. Damit lässt sich praktisch auf jede Aktion im Zusammenhang mit einer Menüauswahl reagieren. In die Ereignisprozedur eines Menüeintrags gelangen Sie wie üblich auf drei verschiedene Arten:

- ➔ durch Doppelklick auf den Eintrag
- ➔ über die Events im Eigenschaftenfenster
- ➔ durch Auswahl des Menüeintrags aus der linken Auswahlliste des Programmcodefensters

Kontextmenüs

Ein Kontextmenü ist ein Menü, das nicht von Anfang an sichtbar ist, sondern in der Regel durch Anklicken eines Steuerelements oder der Innenfläche des Formulars geöffnet wird. Ein Kontextmenü steht in der Toolbox als eigener Eintrag (CONTEXT-MENUSTRIP) zur Verfügung. Ziehen Sie es auf ein Formular, erscheint es zwar unterhalb der Titelleiste und lässt sich wie ein reguläres Menü erstellen, es wird aber anschließend wieder ausgeblendet, da es während der Programmausführung zunächst ebenfalls unsichtbar ist. Möchten Sie es im Entwurfsmodus erneut anzeigen, klicken Sie einfach das Symbol im Komponentenfach unterhalb des Formulars an.

Damit ein Kontextmenü sichtbar wird, muss dessen *Show*-Methode aufgerufen werden, der die Position übergeben wird, an der das Menü erscheinen soll. Da ein Kontextmenü im Allgemeinen über einen Klick mit der *rechten* Maustaste geöffnet wird, erfolgt der Aufruf im *MouseDown*-Event eines Formulars oder eines Steuerelements.



Hier ein kleines Beispiel für die Anzeige des Kontextmenüs ContextMenuStrip1 im MouseDown-Event des Formulars:

```
Private Sub Form1_MouseDown(ByVal sender As Object, ByVal e _
    As System.Windows.Forms.MouseEventArgs) Handles Me.MouseDown
    If e.Button = Windows.Forms.MouseButtons.Right Then
        ContextMenuStrip1.Show(Me.PointToScreen(e.Location), _
            ToolStripDropDownDirection.BelowRight)
    End If
End Sub
```

Achten Sie auf die Umrechnung der Mauskoordinaten über die *PointToScreen*-Methode des Formulars. Da *Show* Bildschirmkoordinaten erwartet, *MouseDown* aber die Koordinaten bezogen auf das Formular liefert, müssen diese entsprechend konvertiert werden. Sehr elegant ist auch die Möglichkeit, über die *Show*-Methode festzulegen, in welche Richtung sich das Menü bezogen auf die Koordinaten öffnen soll. .NET 2.0 überzeugt durch viele dieser netten Kleinigkeiten.

Kontextmenüs direkt zuweisen

Noch ein wenig einfacher geht das Zuordnen eines Kontextmenüs über die Eigenschaften *ContextMenu* bzw. *ContextMenuStrip* eines Formulars oder Steuerelements. In diesem Fall wird das Menü standardmäßig über die rechte Maustaste geöffnet, und Sie müssen den *MouseDown*-Event nicht bemühen:

```
Me.ContextMenuStrip = ContextMenuStrip1
```

Sich selbst zeichnende Menüeinträge


Was bei früheren Visual Basic-Versionen gewisse API-Klimmzüge erforderte und daher in der Regel für Visual Basic-Programmierer nicht infrage kam, lässt sich seit Visual Basic .NET etwas leichter bewerkstelligen. Besitzt die *OwnerDraw*-Eigenschaft eines *MenuItem*-Objekts, das heißt eines einzelnen Menüeintrags, den Wert *True*, wird vor der Anzeige des Eintrags die *DrawItem*-Ereignisprozedur aufgerufen. Dabei besteht die Gelegenheit, im Prinzip beliebige Zeichenoperationen auszuführen. Zwar wird dieser Ereignisprozedur ein *Graphics*-Parameter übergeben, der für die zur Verfügung stehende Zeichenfläche steht, doch ist es leider nicht damit getan, einfach die *DrawIcon*-Methode aufzurufen, um ein Icon zu zeichnen. Der Grund liegt darin, dass die Zeichenfläche auch eine Höhe erhalten muss, die von dem zu zeichnenden Icon und der anzuzeigenden Schriftart abhängt. Da Menüeinträge bei .NET 2.0 nicht nur Bitmaps anzeigen können, sondern sich auch die Schriftarten individuell einstellen lassen, dürfte es für sich selbst zeichnende Menüs nur wenige Notwendigkeiten geben. Ein Beispiel für ein *OwnerDraw*-Control erhalten Sie in Kapitel 8 am Beispiel einer *ListBox*, die jeden Eintrag in einer eigenen Schriftfarbe anzeigt.

9.2 Symbolleisten – das neue ToolStrip-Control

Symbolleisten gehören wie die Menüs einfach zu einer modernen Benutzeroberfläche. Leider hatten gerade Visual Basic-Programmierer in den letzten Jahren nicht allzu viel Spaß mit Symbolleisten. Das ist spätestens mit Visual Basic 2005 vorbei, denn das neue *ToolStrip*-Control präsentiert sich vielseitig und vor allem leicht bedienbar. Ordnen Sie es auf einem Formular an, erhalten Sie davon einen ersten Vorgesmack, denn es zeigt sich nicht nur in einer hübschen und modernen Optik, sondern bietet eine Auswahlliste, über die Sie Elemente zur Symbolleiste hinzufügen. Das Repertoire ist wirklich beeindruckend, denn neben einem Button stehen als weitere Elemente zur Verfügung: Label, SplitButton, DropDownButton, ComboBox, TextBox und der obligatorische Separator. Damit lassen sich wirklich funktionale Symbolleisten erstellen. Fällt Ihnen der kleine Pfeil in der rechten oberen Ecke auf? Nach einem Klick darauf erscheint das übliche Auswahlmenü, das besonders wichtige Aktionen zusammenfasst. Rufen Sie den Eintrag STANDARD-ELEMENTE EINFÜGEN auf, und lassen Sie sich wieder einmal angenehm überraschen – fertig ist die Symbolleiste mit allen wichtigen Elementen, die ein Anwendungsfenster heutzutage braucht. Die Symbolleiste können Sie selbstverständlich beliebig editieren.

Abbildung 9.5:


Wird auf Knopfdruck angelegt – eine Symbolleiste mit Standardelementen.



Weder das ToolStrip- noch das MenuStrip-Control muss am oberen Fensterrand andocken, es stehen verschiedene Dockpositionen zur Auswahl.

Abbildung 9.6:

Im Auflistungs-Editor kann jedes Element der Toolleiste nachträglich bearbeitet werden.



Beispielprojekt Aktiendepotverwaltung

Wer vom (praktisch permanent stattfindenden) Börsenboom profitieren möchte, benötigt als Erstes eine Depotverwaltung (und als Nächstes einen guten Steuerberater). Auch wenn es das natürlich alles online gibt, Selbermachen bereitet zum einen mehr Spaß und bietet zum anderen die Gelegenheit, eine maßgeschneiderte Lösung zu erhalten. Die Aktiendepotverwaltung auf der Buch-CD soll weniger Ihr Vermögen mehren (und vor allem niemanden zu riskanten Spekulationen verleiten), sondern in erster Linie (eigentlich sogar ausschließlich) ein nettes Beispiel für die verschiedenen Windows Forms-Steuer-elemente bieten.



Sie finden das Beispiel auf der Buch-CD in der Projektdatei 09_Aktiendepot.sln.



Abbildung 9.7: Eine kleine Depotverwaltung (alle Angaben natürlich ohne Gewähr)

Die Rolle des ToolStrip-Containers

Anwender einer modernen Office-Anwendung sind es gewohnt, dass sie die Befehlsleisten beliebig im Anwendungsfenster verschieben können. Damit dies auch unter einer Windows Forms-Anwendung realisiert werden kann, wurde mit .NET 2.0 das ToolStripContainer-Control eingeführt. Es stellt eine während der Programmausführung unsichtbare Fläche dar, innerhalb derer die verschiedenen Strip-Controls an den vier Rändern des Containers angedockt werden können. Neben der zentralen Fläche in der Mitte, repräsentiert durch ein *ContentPanel*, wird jeder Rand durch ein eigenes Panel-Control (etwa *TopToolStripPanel*) repräsentiert. Um ein ToolStrip-Control in einen Container einzubetten, ist es am einfachsten, aus dessen Aufgabenmenü den Eintrag `IN TOOLSTRIPCONTAINER EINBETTEN` aufzurufen, woraufhin dann der Container nachträglich hinzugefügt wird.





Abbildung 9.8: Ein ToolStrip-Control wird nachträglich in einen ToolStripContainer eingebettet.

Der Umgang mit dem ToolStrip-Container ist am Anfang ein wenig gewöhnungsbedürftig, da sich das Control gegen das Selektieren mit der Maus ein wenig sträubt und es nicht ganz einfach ist, ein hinzugefügtes *ToolStripContainer*-Control wieder zu entfernen (es muss dazu nicht nur leer sein, es darf sich auch kein anderes *ToolStripContainer*-Control mit einem Inhalt auf dem Formular befinden). Auch das Herausnehmen eines Controls aus dem »Einflussbereich« eines Containers scheint nicht sehr gut zu funktionieren. Dazu ein Tipp, der schnell übersehen wird: Klicken Sie die Innenfläche eines Steuerelements mit der rechten Maustaste an, und wählen Sie zunächst `AUSWÄHLEN` und anschließend das gewünschte Control. Dadurch lässt sich ein Control in der Regel am einfachsten selektieren.

Abbildung 9.9:

Die Symbolleiste kann dank des ToolStripContainers flexibel angedockt werden.



Unter Umständen ist es einfacher, den Designer zu ignorieren und eine Toolleiste inklusive des umgebenden Containers im Code anzuordnen:

```
Dim toolStripContainer1 = New ToolStripContainer()
Dim toolStrip1 = New System.Windows.Forms.ToolStrip()
toolStrip1.Items.Add("1")
toolStrip1.Items.Add("2")
toolStrip1.Items.Add("3")
toolStripContainer1.TopToolStripPanel.Controls.Add(toolStrip1)
Controls.Add(toolStripContainer1)
```

Diese Befehle legen eine Toolleiste mit drei Einträgen an, die in einen Container eingebettet wird.

9.3 Speziellere Steuerelemente

In diesem Abschnitt werden die etwas spezielleren Steuerelemente vorgestellt, die Sie am Anfang seltener benötigen werden, die aber ebenfalls zu den Stammsteuerelementen gehören.

Die StatusStrip-Klasse

Das ebenfalls neu mit .NET 2.0 eingeführte StatusStrip-Control ist für die Statusleiste zuständig, die jedem Formular zugeordnet werden kann. Anders als in der Vergangenheit, sprich in der »VB6-Ära«, bietet die Statusleiste einiges an Möglichkeiten:

- ➔ Neben einer simplen TextBox stehen auch ein Fortschrittsbalken und eine DropDown-Liste zur Auswahl.
- ➔ Es können auch Hyperlinks angeordnet werden.
- ➔ Es kann dem Anwender erlaubt werden, die Anordnung der Elemente während der Programmausführung bei gedrückter **[Alt]**-Taste zu ändern. Dazu ist die *AllowItemReorder*-Eigenschaft auf *True* zu setzen.

Insbesondere die Möglichkeit, Auswahllisten, Fortschrittsbalken und Links anzuzugreifen, macht die neue Statusbar sehr attraktiv.

Das Bearbeiten des StatusStrip-Controls und der dazugehörigen Einträge geschieht am besten über den Elementauflistungs-Editor.



Eigenschaft	Bedeutung
<i>AllowItemReorder</i>	<i>True</i> , wenn die Elemente vom Anwender in der Reihenfolge vertauscht werden können.
<i>LayoutStyle</i>	Ermöglicht es, die Elemente nicht nur nebeneinander, sondern zum Beispiel auch überlappend darzustellen. Die beste Einstellung ist <i>Table</i> , hier besitzt jedes Element seinen festen Bereich. Bei <i>Flow</i> folgen die Elemente, je nach aktueller Breite, unmittelbar aufeinander.
<i>TextDirection</i>	Legt die Ausrichtung des Textes fest, wobei der Text auch um 90 oder 270 Grad gedreht, also vertikal angezeigt werden kann.

Tabelle 9.2:
Ein paar interessante Eigenschaften der StatusStrip-Klasse

Die optimale Breite der einzelnen Elemente festzulegen ist nicht ganz einfach. Durch Setzen der Spring-Eigenschaft der ToolStripStatusLabel-Klasse auf True kann erreicht werden, dass ein Element den noch verbleibenden Platz komplett ausfüllt.



Anordnen eines Fortschrittsbalkens in der Statusleiste

Im Folgenden soll an einem kleinen Beispiel nicht nur gezeigt werden, wie sich (größere) Dateien asynchron laden lassen, sondern auch, wie der aktuelle Stand beim Laden per Fortschrittsbalken in der Statusleiste visualisiert werden kann.

Sie finden das Beispiel auf der Buch-CD in der Projektdatei 09_Progressbar.sln.



Beispiel

Das folgende Beispiel durchsucht ein Verzeichnis mit aus Outlook (oder einem anderen E-Mail-Programm) heraus gespeicherten E-Mail-Nachrichten Zeile für Zeile und zeigt sowohl den Namen der gerade durchsuchten Datei, die Zeile sowie den Fortschritt mithilfe zweier Labels (tslbDateiname und tslbDateizeile) und eines Fortschrittsbalkens (tspbDateien) in der Statusbar an. Damit bei wenigen Dateien überhaupt etwas zu sehen ist, wird nach jeder Zuweisung die Refresh-Methode aufgerufen und die Programmausführung durch die Threading.Thread.Sleep-Methode ein wenig »abgebremst«.



```
With tspbDateien
    .Minimum = 0
    .Maximum = New DirectoryInfo(MsgPfad).GetFiles("*.txt").Length
End With
```

```

For Each Fi As FileInfo In New DirectoryInfo(MsgPfad).GetFiles("*.txt")
    AnzahlDateien += 1
    AnzahlZeilen = 0
    tspbDateien.Value = AnzahlDateien
    ts1bDateiname.Text = Fi.Name.Substring(0, Math.Min(Fi.Name.Length, _
        40))
    StatusStrip1.Refresh()
    Threading.Thread.Sleep(1000)
    Dim Sr As New StreamReader(Fi.FullName)
    Do While Sr.Peek <> -1
        Dim Zeile As String = Sr.ReadLine
        ' Hier müsste die Zeile untersucht werden
        AnzahlZeilen += 1
        ts1bDateizeile.Text = AnzahlZeilen
        StatusStrip1.Refresh()
    Loop
    Sr.Close()
    Threading.Thread.Sleep(1000)
Next
Threading.Thread.Sleep(1000)
tspbDateien.Value = 0

```

Abbildung 9.10:

Die Statusleiste zeigt zwei Labels und einen Fortschrittsbalken an.



Daten in einem Baum anordnen – das TreeView-Steuerelement

Das `TreeView`-Control zeigt einen aus Knoten bestehenden Baum von Textelementen an, wobei jedem Knoten eine kleine Bitmap zugeordnet werden kann. Jeder Knoten wird durch ein `TreeNode`-Objekt repräsentiert. Es besitzt selbst eine `Nodes`-Auflistung, an die weitere `TreeNode`-Objekte gehängt werden können, sodass sich eine theoretisch beliebig tief verschachtelte Baumstruktur realisieren lässt. Ein `TreeNode`-Objekt besteht aus einem Text und einem Icon. Letzteres kann leider nicht direkt zugeordnet werden, sondern muss aus einer `ImageList` stammen. Die Zuordnung geschieht über einen Index oder ein Schlüsselwort. Über verschiedene Eigenschaften der `TreeNode`-Klasse kommt man schnell an die Nachbarknoten heran. Der Umgang mit dem `TreeView`-Steuerelement ist relativ einfach, wie das folgende Beispiel zeigt.

Ein neuer Knoten wird wie folgt angelegt:

```
Dim Tn As TreeNode = TreeView1.Nodes.Add("Neuer Eintrag")
```

Die Variable `Tn` steht für den neuen Knoten, sodass der folgende Befehl einen Unterknoten anhängt:

```
Dim SubTn As TreeNode = Tn.Nodes.Add("Ein Untereintrag")
```

Soll der Unterknoten nicht weiter angesprochen werden, muss der Rückgabewert von *Add* natürlich keiner Variablen zugewiesen werden.

Sie finden das Beispiel auf der Buch-CD in der Projektdatei 09_TreeViewBsp.sln.



Das Beispiel listet alle Dateien mit der Erweiterung *.doc* im Verzeichnis Eigene Dateien als Knoten auf und fügt verschiedene Dokumenteigenschaften als Unterknoten hinzu. Jeder Dokumentknoten wird durch ein kleines Icon dargestellt, das aus einem *ImageList-Control* stammt. Anders als die mit *.NET 2.0* eingeführten *Strip-Controls* ist dies die einzige Möglichkeit, um einem *TreeView-* oder einem *ListView-Control* Icons zuzuordnen.



```
Dim DocPfad As String = Environment.GetFolderPath(Environment.SpecialFolder.MyDocuments)
Dim Dso As Object = CreateObject("DSOLEFile.PropertyReader")
Dim DsoProps As Object
For Each Fi As FileInfo In New DirectoryInfo(DocPfad).GetFiles("*.doc")
    DsoProps = Dso.GetDocumentProperties(Fi.FullName)
    Dim Tn As TreeNode = TreeView1.Nodes.Add(Fi.Name, Fi.Name, 0)
    Tn.Nodes.Add("Autor", "Autor: " & DsoProps.Author, 1)
    Tn.Nodes.Add("AnzahlWorte", "Anzahl Worte: " & DsoProps.WordCount, 1)
    Tn.Nodes.Add("Dokumentversion", "Dokumentversion: " & _
        DsoProps.RevisionNumber, 1)
Next
```




Abbildung 9.11: Das *TreeView-*Control zeigt ein Baumstruktur an.

Kurzer Exkurs: Eigenschaften in Office-Dokumenten lesen

Wie lassen sich die zahlreichen Eigenschaften eines Microsoft Office-Dokuments wie der Name des Autors, die Anzahl der Seiten, das Datum der letzten Änderung usw. lesen? Entweder indem ein Programm per *COM-Interop* auf die jeweilige

Office-Anwendung zugreift, die das Dokument lädt und die einzelnen Eigenschaften über die *BuiltInDocumentProperties*-Eigenschaft ausliest. Oder über eine kleine COM-Dll mit dem Namen *DsoFile.dll*, die von der Microsoft-Website heruntergeladen und registriert werden muss, sodass sie jeder Anwendung zur Verfügung steht.



Das folgende Beispiel liest die Author-Eigenschaft eines Word-Dokuments, das über das FileInfo-Objekt Fi angesprochen wird.

```
Dim Dso As Object = CreateObject("DSOFile.PropertyReader")
Dim DsoProps As Object
DsoProps = Dso.GetDocumentProperties(Fi.FullName)
Autor = DsoProps.Author
```

Anstelle von CreateObject kann auch über PROJEKT/VERWEIS HINZUFÜGEN ein Verweis auf die Datei Dsofile.dll eingebunden und die Variable wie folgt instanziiert werden :

```
Dim Dso As New DSOFile.PropertyReader
```



Mehr zu dem Thema finden Sie in dem Knowledgebase-Artikel <http://www.microsoft.com/germany/technet/datenbank/articles/600696.msp>.

Wie erhält ein TreeView-Knoten sein Icon?

Voraussetzung dafür, dass einem TreeView-Knoten ein Icon zugeordnet werden kann, ist eine ImageList-Komponente, die ebenfalls über die Toolbox angeboten wird. Die einzige Aufgabe einer ImageList-Komponente ist es, anderen Steuerelementen (im Wesentlichen nur ListView und TreeView, weil das Toolbar-Control von .NET 1.1 in .NET 2.0 durch das ToolStrip-Control abgelöst wurde) Bitmaps zur Verfügung zu stellen. Da die ImageList zur Laufzeit unsichtbar ist, ist sie als Komponente und nicht als Control implementiert und wird nach dem Anordnen auf dem Formular im »Komponentenfach« angezeigt. Der Umgang mit der ImageList besteht aus folgenden Schritten:



1. Hinzufügen von Bitmaps über die *Images*-Eigenschaft. Dabei kommt es vor allem auf die richtige Größe an. Auch wenn die ImageList in diesem Punkt nicht wählerisch ist, eine große Bitmap mit einer Auflösung von zum Beispiel 640 × 480 Pixel oder höher sieht in einem TreeView im Allgemeinen nicht besonders vorteilhaft aus, wenn sie auf 32 × 32 oder 16 × 16 Pixel herunterskaliert wurde. Sehr praktisch ist, dass die Größe der Bitmap, die einem angeschlossenen Steuerelement zur Verfügung gestellt wird, in der Aufgabenliste eingestellt werden kann.
2. Dieser Schritt ist optional: das Festlegen einer transparenten Farbe. Viele Icons, insbesondere jene, die Teil von Visual Studio 2005 sind, enthalten eine Hintergrundfarbe (zum Beispiel Lila, genauer gesagt Fuchsia mit dem RGB-Code 255, 0, 255), die über die Eigenschaft *TransparentColor* als transparente Farbe ausgewählt werden muss, damit der Hintergrund entsprechend durchsichtig wird. Wenn die Hintergrundfarbe nicht bekannt ist, bietet es sich an, diese mit einem Bildbearbeitungsprogramm zu ermitteln, da ein Erraten des richtigen Farbcodes nur schwerlich möglich ist.

3. Die ImageList wird dem ListView- oder TreeView-Steuerelement über deren Eigenschaften *LargeImageList* und *SmallImageList* bzw. *ImageList* zugeordnet.
4. Das Icon wird über seine Indexnummer angesprochen:

```
Dim Tn As TreeNode = TreeView1.Nodes.Add(Fi.Name, Fi.Name, 0)
```

Es ist kein Problem, Bitmaps nachträglich zu ändern, wenn die ImageList bereits einem Control zugeordnet wurde.



Auf Ereignisse des TreeView reagieren

Ein NodeSelect-Ereignis gibt es beim TreeView-Control nicht. Das Ereignis, das diesem am nächsten kommt, ist der *AfterSelect*-Event, der ausgelöst wird, nachdem ein Knoten selektiert wurde. Über den *TreeViewEventArgs*-Parameter wird der selektierte Knoten übergeben.

Oft soll eine Aktion nur für Knoten ausgeführt werden, die sich eine Ebene höher in der Knotenhierarchie befinden. In diesem Fall lässt sich über die Parent-Eigenschaft feststellen, ob der Knoten einen »Elternknoten« besitzt oder, sollte das nicht der Fall sein, ein oberster Knoten ist. Sehr praktisch ist auch die universelle Tag-Eigenschaft, der beliebige Werte zugewiesen werden können, unter anderem eine Information, die etwas über die Position des Knotens aussagt. Eine Level-Eigenschaft gibt es bei der TreeNode-Klasse auch.



Mitglied	Bedeutung
<i>CheckBoxes</i> -Eigenschaft	Legt fest, ob jeder Knoten durch eine CheckBox dargestellt wird. Sehr interessante Eigenschaft.
<i>CollapseAll</i> -Methode	Klappt alle Knoten auf einmal zu.
<i>ExpandAll</i> -Methode	Klappt alle Knoten auf einmal auf.
<i>GetNodeCount</i> -Eigenschaft	Gibt die Anzahl der Knoten des aktuellen Knotens zurück.
<i>HideSelection</i> -Eigenschaft	Legt fest, ob die aktuelle Auswahl erhalten bleibt, wenn das Steuerelement den Fokus verliert.
<i>HotTracking</i> -Eigenschaft	Legt fest, ob ein Knotenname als Hyperlink dargestellt wird, wenn der Mauszeiger darauf positioniert wird. Sehr interessante Eigenschaft.
<i>ImageIndex</i> -Eigenschaft	Steht für den Index des mit dem Knoten angezeigten Icons und bezieht sich auf ein ImageList-Steuerelement.
<i>ImageKey</i> -Eigenschaft	Erlaubt das einem Knoten zugeordnete Icon über einen Schlüsselnamen auszuwählen.
<i>ImageList</i> -Eigenschaft	Legt das ImageView-Steuerelement fest, das die Icons zur Verfügung stellt.

Tabelle 9.3:
Interessante
Mitglieder der
TreeView-Klasse

Tabelle 9.3:
Interessante
Mitglieder der
TreeView-Klasse
(Forts.)

Mitglied	Bedeutung
<i>LabelEdit</i> -Eigenschaft	<i>True</i> , wenn der Knotenname bearbeitet werden kann.
<i>Nodes</i> -Eigenschaft	Steht für die Unterknoten des Knotens als <i>Nodes</i> -Auflistung.
<i>SelectedNode</i> -Eigenschaft	Steht für den aktuell selektierten Knoten.
<i>Sort</i> -Methode	Sortiert die Knotennamen alphabetisch.
<i>TopNode</i> -Eigenschaft	Steht für den ersten sichtbaren Knoten im TreeView.
<i>TreeNodeSorter</i> -Eigenschaft	Steht für einen Vergleichler, der <i>IComparer</i> implementiert, um die Knotenelemente sortieren zu können.

Tabelle 9.4:
Interessante
Mitglieder der
TreeNode-Klasse

Mitglied	Bedeutung
<i>Checked</i> -Eigenschaft	<i>True</i> , wenn das Element selektiert ist (wenn die <i>CheckBoxes</i> -Eigenschaft des TreeView entsprechend gesetzt ist).
<i>Collapse</i> -Methode	Klappt alle Unterknoten des Knotens zu. Per <i>True</i> -Parameter behalten die Kindknoten ihren aktuellen Zustand.
<i>EnsureVisible</i> -Methode	Zeigt den Baum so an, dass dieser Knoten sichtbar ist.
<i>ExpandAll</i> -Methode	Erweitert den Knoten, sodass alle Kindknoten sichtbar werden.
<i>FirstNode</i> -Eigenschaft	Steht für den ersten Unterknoten.
<i>FullPath</i> -Eigenschaft	Kompletter Pfad des Knotens.
<i>GetNodeCount</i> -Methode	Gibt die Anzahl der Unterknoten zurück.
<i>IsSelected</i> -Eigenschaft	<i>True</i> , wenn der Knoten aktuell selektiert ist.
<i>LastNode</i> -Eigenschaft	Steht für den letzten Unterknoten.
<i>Level</i> -Eigenschaft	Steht für die Ebene des Knotens innerhalb der Knotenhierarchie.
<i>Nodes</i> -Eigenschaft	Steht für die Unterknoten als <i>Nodes</i> -Auflistung.
<i>Parent</i> -Eigenschaft	Steht für den übergeordneten Knoten.
<i>Remove</i> -Methode	Entfernt den Knoten aus der <i>Nodes</i> -Auflistung.
<i>Tag</i> -Eigenschaft	Allgemeine Eigenschaft vom Typ <i>Object</i> , durch die jedem Knoten etwas Beliebiges, etwa eine ID, zugeordnet werden kann.
<i>Text</i> -Eigenschaft	Der angezeigte Text des Knotens.
<i>Toggle</i> -Methode	Schaltet den Knoten zwischen erweitert und reduziert um.
<i>TreeView</i> -Eigenschaft	Steht für das TreeView-Control, zu dem der Knoten gehört.

Ein Formular beliebig unterteilen – das TabControl-Steuerelement

Ein TabControl-Steuerelement ist eine sehr praktische Angelegenheit, denn es stellt eine beliebige Anzahl von Unterteilungen, auch Registerkarten oder Tabs genannt, zur Verfügung, auf denen Steuerelemente angeordnet werden können. Jede Registerkartenseite wird durch ein *TabPage*-Objekt repräsentiert. Benötigt man eine typische Registerkarte, um auf konstantem Raum prinzipiell beliebig viele Eingabelemente – unterteilt nach Kategorien – unterzubringen, ist das TabControl dafür bestens geeignet. Auch wenn sich das TabControl, wie jedes Steuerelement, komplett programmgesteuert einrichten lässt, ist es am einfachsten, das Anlegen der Register im Eigenschaftfenster vorzunehmen.




Abbildung 9.12: Ein TabControl erlaubt das Unterteilen eines Formulars in Register (Tabs).

Mitglied	Bedeutung
<i>DisplayRectangle</i> -Eigenschaft	Steht für die Innenfläche der stets gleich großen Registerseite als <i>Rectangle</i> -Struktur.
<i>Multiline</i> -Eigenschaft	<i>True</i> , wenn sich die Reiter auf mehrere Zeilen verteilen können – ansonsten wird ein Pfeil zum Scrollen angezeigt.
<i>SelectedTab</i> -Eigenschaft	Repräsentiert die momentan ausgewählte Registerkarte als <i>TabPage</i> -Objekt.
<i>TabCount</i> -Eigenschaft	Anzahl der Registerkarten.
<i>TabPage</i> -Eigenschaft	Steht für die Auflistung der Registerseiten in Gestalt einer <i>TabPageCollection</i> , die entsprechende Methoden wie <i>Add</i> , <i>Insert</i> und <i>Remove</i> anbietet.

Tabelle 9.5: Interessante Mitglieder der TabControl-Klasse

Das TabControl-Steuerelement besitzt leider auch in der aktuellen Version 2.0 einen lästigen Bug, der darin besteht, dass manchmal die Register ihre Anordnung ändern. Sie müssen dann in der Designer-Datei des Formulars die Stelle lokalisieren, an der die Seiten über die Add-Methode der Controls-Auflistung zum TabControl hinzugefügt werden, und die Reihenfolge an dieser Stelle wieder korrigieren. Dadurch wird das Problem aber nur vorübergehend behoben und nicht ganz abgestellt.



Steuerelemente mit der Find-Methode der Controls-Auflistung finden

Je umfangreicher ein Formular wird, desto nützlicher kann es sein, einzelne Steuerelemente über den Namen lokalisieren zu können. Stellen Sie sich ein TabControl mit zwölf Registern vor, auf dem jedes Register eine ComboBox enthält, die Monate anzeigt. Theoretisch muss die Liste zwölf Mal gefüllt werden. Statt zwölf Aufrufen wäre genau ein Aufruf sinnvoller, dem einfach das zu füllende Control übergeben wird. Soll das Füllen erst dann geschehen, wenn das Register aktiviert wird, muss der Füllroutine jene ComboBox übergeben werden, die sich auf dem derzeit selektierten Register befindet. Da dieses Register unter Umständen mehrere Steuerelemente und auch mehrere ComboBoxen enthält, wäre eine Suchfunktion sehr praktisch. Diese existiert in Gestalt der *Find*-Methode der *Controls*-Auflistung. Diese Methode lokalisiert ein Control anhand des dazugehörigen Namens und gibt das Control zurück.



Sie finden das Beispiel auf der Buch-CD in der Projektdatei 09_TabFindControl.sln.



Das folgende Beispiel gibt die (einzige) ComboBox zurück, die sich auf der momentan selektierten Registerkarte im TabControl tbMaerkte befindet, wobei lediglich der Name der ComboBox bekannt ist. Wurde diese gefunden, wird sie einer Prozedur als Parameter übergeben:

```
Try
    Dim cbTemp As Control = tbMaerkte.SelectedTab.Controls.Find _
        ("cbMonateRH" & CStr(tbMaerkte.SelectedTab.Tag), False)(0)
    MonatsListeFuellen(CType(CbTemp, ComboBox))
Catch ex As SystemException
    ' Hier passiert nichts
End Try
```

Ist der Name nicht bekannt, kann die *Controls*-Auflistung auch per *For Each* durchsucht werden – und sollten Controls vorhanden sein, die wiederum andere Controls als Container enthalten können, muss die Suche rekursiv erfolgen, was aber nicht weiter kompliziert ist, da dazu bei *Find* lediglich ein *True*-Flag übergeben werden muss.

Die Explorer-Ansicht für beliebige Gegenstände – das ListView-Steuerelement

Das ListView-Steuerelement ist ein alter Bekannter, da es zu den »Windows Common Controls« gehört, die (seit Windows 95) vom Betriebssystem zur Verfügung gestellt werden. Es stellt die typische Innenansicht des Windows-Explorers zur Verfügung, bei der die Elemente wahlweise als große Symbole, als kleine Symbole, in Form einer Listenansicht oder als Detailansicht angezeigt werden können. Eine ListView ist immer dann sehr praktisch, wenn eine mehrspaltige Liste benötigt wird. Da das ListView-Control von .NET 2.0 lediglich das Standardsteuerelement

der Windows-Library kapselt, unterscheidet sich die Programmierung unter Visual Basic 2005 nur geringfügig von der unter VB6. Auch wenn eine ListView insgesamt fünf Ansichten bietet (Details, große Symbole, kleine Symbole, Liste und geschachtelt), spielt für die Praxis meistens nur die Detailansicht eine Rolle, die durch Setzen der *View*-Eigenschaft auf den Wert *Details* aktiviert wird.

Eine ListView erhält ihren Inhalt in Gestalt von Texten, die wie bei einer ListBox über die *Add*-Methode der *Items*-Auflistung hinzugefügt werden. Anders als eine ListBox kann jeder Eintrag Untereinträge besitzen, die über die *Add*-Methode der *SubItems*-Eigenschaft eines *ListViewItem*-Objekts hinzugefügt werden, das von der *Add*-Methode der *Items*-Auflistung zurückgegeben wird. Voraussetzung ist allerdings, dass zuvor eine Spalte zur *Columns*-Auflistung der ListView hinzugefügt wurde.

Sie finden das Beispiel auf der Buch-CD in der Projektdatei 09_ListViewBsp.sln.



Das folgende Beispiel listet MP3-Musikdateien im Verzeichnis Eigene Musik in einem ListView-Control auf und gibt zusätzlich zur Größe auch die Bitrate und die Länge der Datei an. Da dies allerdings mit den Mitteln der .NET-Klassenbibliothek alleine nicht möglich ist, greift das Programm auf eine Komponente zurück, welche die so genannten ID3-Tags einer MP3-Datei ausliest. Eine solche Komponente ließe sich mit relativ wenig Aufwand selbst programmieren (mehr dazu in Kapitel 11, wenn es um eigene Komponenten geht), Sie finden aber auch viele fertige Lösungen im Internet – das folgende Beispiel bedient sich eines Projektes von Thommy Mewes, das Sie unter <http://www.codeproject.com/vb/net/vbmp3header.asp> finden.



```
Dim MusikPfad As String = Environment.GetFolderPath _
    (Environment.SpecialFolder.MyMusic)
Dim Mp3Info As New Mp3.MP3Info
' ListView einrichten
ListView1.Columns.Add("Name", 360)
ListView1.Columns.Add("Dauer", 80)
ListView1.Columns.Add("Bitrate", 80)
ListView1.Columns.Add("Abtastrate", 80)
ListView1.View = View.Details
For Each Fi As FileInfo In New DirectoryInfo(MusikPfad). _
    GetFiles("*.mp3")
    Dim Lv As ListViewItem = ListView1.Items.Add(Fi.Name)
    Mp3Info.FileName = Fi.FullName
    Dim T As New TimeSpan(0, 0, Mp3Info.Length)
    Lv.SubItems.Add(T.ToString)
    Lv.SubItems.Add(Mp3Info.Bitrate)
    Lv.SubItems.Add(Mp3Info.SamplingRateFrequency)
Next
```

Abbildung 9.13:
Das ListView-
Control in Aktion




Tabelle 9.6:
Interessante
Mitglieder der
ListView-Klasse

Mitglied	Bedeutung
<i>AllColumnReorder</i> -Eigenschaft	Wenn <i>True</i> , darf der Anwender die Anordnung der Spalten verschieben.
<i>CheckBoxes</i> -Eigenschaft	Wenn <i>True</i> , wird neben jedem Eintrag eine <i>CheckBox</i> angezeigt.
<i>CheckedIndices</i> -Eigenschaft	Steht für eine <i>CheckedIndexCollection</i> , welche die Indizes aller selektierten Elemente enthält.
<i>CheckedItems</i> -Eigenschaft	Steht für eine <i>CheckedListViewItemCollection</i> , die alle selektierten <i>ListView</i> -Elemente enthält.
<i>Clear</i> -Methode	Entfernt alle Elemente und Spalten.
<i>Columns</i> -Eigenschaft	Steht für die Auflistung aller Spalten in Gestalt einer <i>ColumnHeaderCollection</i> .
<i>FindItemWithText</i> -Methode	Lokalisiert das erste <i>ListViewItem</i> -Objekt mit dem angegebenen Textwert.
<i>FocusedItem</i> -Eigenschaft	Steht für das aktuell selektierte <i>ListViewItem</i> -Objekt.
<i>FullRowSelect</i> -Methode	Wenn <i>True</i> , wird beim Anklicken eines Eintrags die gesamte Reihe selektiert.
<i>GetItemAt</i> -Methode	Lokalisiert ein <i>ListView</i> -Element anhand seiner Position.
<i>HeaderStyle</i> -Eigenschaft	Legt fest, ob der Spaltenkopf anklickbar ist oder nicht.
<i>Items</i> -Eigenschaft	Steht für alle Elemente in Gestalt einer <i>ListViewItemCollection</i> .
<i>LabelEdit</i> -Eigenschaft	Wenn <i>True</i> , kann der Inhalt eines Elementes (durch erneutes Anklicken) editiert werden.
<i>ListViewItemSorter</i> -Eigenschaft	Erlaubt das Festlegen eines Vergleichersobjekts, das <i>IComparer</i> implementiert.

Mitglied	Bedeutung
<i>Scrollable</i> -Eigenschaft	Wenn <i>True</i> , wird eine Bildlaufleiste angezeigt.
<i>SelectedItem</i> -Eigenschaft	Steht für die aktuell selektierten Elemente in Gestalt einer <i>SelectedListViewItemCollection</i> .
<i>Sort</i> -Methode	Sortiert die Einträge nach der ersten Spalte alphabetisch (dies ist der normalerweise benötigte Fall).
<i>Sorting</i> -Eigenschaft	Legt fest, ob aufsteigend oder absteigend sortiert wird.
<i>View</i> -Eigenschaft	Legt den Anzeigemodus des ListView-Controls fest.

Tabelle 9.6:
Interessante Mitglieder der *ListView*-Klasse (Forts.)

Datum komfortabel auswählen – das *MonthCalendar*-Steuerelement

Für die komfortable Auswahl eines Datums gibt es gleich zwei Steuerelement: *MonthCalendar* und *DateTimePicker*. Während Ersteres ein relativ attraktives Kalenderblatt darstellt, ist Letzteres eine etwas überdimensionierte *TextBox*, die ein Datum anzeigt und bei der sich das Kalenderblatt nur auf Mausklick öffnet. Ein wenig irritierend ist, dass es scheinbar keine Eigenschaft gibt, die für das ausgewählte Datum steht. Das »Geheimnis« besteht darin, dass stets ein Datumsbereich markiert werden kann und das selektierte Datum daher über die *SelectionStart*-Eigenschaft abgefragt wird, die für einen *Date*-Wert steht.




Abbildung 9.14:
Die Datumsauswahl in einer Form, wie man sie von Windows gewohnt ist

Mitglied	Bedeutung
<i>BoldedDates</i> -Eigenschaft	<i>Date</i> -Array mit Datumswerten, die fett hervorgehoben werden sollen.
<i>MaxSelectionCount</i> -Eigenschaft	Maximale Anzahl an auswählbaren Tagen (maximal 7).
<i>SelectionEnd</i> -Eigenschaft	Das Enddatum eines Datumsbereichs.

Tabelle 9.7:
Interessante Mitglieder der *MonthCalendar*-Klasse

Tabelle 9.7:
Interessante Mitglieder der MonthCalendar-Klasse (Forts.)

Mitglied	Bedeutung
<i>SelectionStart</i> -Eigenschaft	Das Startdatum eines Datumsbereichs.
<i>SelectionRange</i> -Eigenschaft	Steht für den ausgewählten Datumsbereich als <i>SelectionRange</i> -Objekt.
<i>SetDate</i> -Methode	Legt ein Datum fest, das selektiert werden soll.
<i>SetSelectionRange</i> -Methode	Legt einen Datumsbereich fest, der selektiert werden soll.
<i>ShowToday</i> -Eigenschaft	<i>True</i> , wenn das aktuelle Datum hervorgehoben werden soll.
<i>ShowTodayCircle</i> -Eigenschaft	<i>True</i> , wenn das aktuelle Datum mit einem Kreis hervorgehoben werden soll.
<i>ShowWeekNumbers</i> -Eigenschaft	<i>True</i> , wenn auch die Wochennummer angezeigt werden soll.
<i>TodayDate</i> -Eigenschaft	Steht für das aktuelle Datum des Kalenders (es muss also nicht immer automatisch das aktuelle Datum sein).

Variable Bereiche mit dem Splitter-Steuererelement und dem SplitContainer

Mithilfe des SplitContainers-Controls erhalten Sie eine Fläche, deren Bereiche beliebig verschoben werden können, so wie es zum Beispiel bei einem Ordnerfenster in der Explorer-Ansicht üblich ist. Der Umgang mit dem *SplitContainer*-Control ist vollkommen selbsterklärend. Nachdem das Control auf einem Formular platziert wurde, können auf den beiden Panel-Flächen Steuerelemente angeordnet werden. Lediglich das Feinjustieren der Elemente ist etwas knifflig. Hier hilft es, sich daran zu erinnern, dass jedes Container-Control am einfachsten durch Anklicken mit der rechten Maustaste und Aufrufen des entsprechenden <ELEMENT> AUSWÄHLEN-Eintrags markiert wird.

Tabelle 9.8:
Interessante Mitglieder der SplitContainer-Klasse

Mitglied	Bedeutung
<i>IsSplitterFixed</i> -Eigenschaft	Legt fest, ob der Splitter verschiebbar sein soll, bzw. gibt den aktuellen Zustand zurück. <i>False</i> bedeutet Splitter ist verschiebbar.
<i>Orientation</i> -Eigenschaft	Legt fest, ob der Splitter horizontal oder vertikal verschoben werden soll.
<i>Panel1</i> -Eigenschaft	Steht für den linken bzw. oberen Bereich in Gestalt eines <i>SplitterPanel</i> -Objekts.
<i>Panel1MinSize</i> -Eigenschaft	Mindestgröße des linken bzw. oberen Bereichs.
<i>Panel2</i> -Eigenschaft	Steht für den rechten bzw. unteren Bereich in Gestalt eines <i>SplitterPanel</i> -Objekts.
<i>Panel2MinSize</i> -Eigenschaft	Mindestgröße des rechten bzw. unteren Bereichs.

Mitglied	Bedeutung
<i>ProcessTabKey</i> -Eigenschaft	Wählt das nächste Steuerelement aus (über <i>True</i> wird die Vorwärts-Richtung eingestellt).
<i>SplitterDistance</i> -Eigenschaft	Position des Splitters in Relation zum linken oder oberen Rand.
<i>SplitterIncrement</i> -Eigenschaft	Legt fest, mit welcher Schrittweite sich der Splitter bewegt.
<i>SplitterWidth</i> -Eigenschaft	Breite des Splitters.

Tabelle 9.8:
Interessante Mitglieder der Split-Container-Klasse (Forts.)

Nur bestimmte Eingaben zulassen – die MaskedTextBox

Eine maskierbare TextBox regelt die Eingabe anhand einer Maske, sodass die Eingabe nur exakt nach einem Muster erfolgen kann, etwa einer internationalen Telefonnummer oder einer Kreditkartennummer. Ein solches Steuerelement gab es bereits vor »Urzeiten« (genauer bei der Version 3.0 von Visual Basic). Da dieses nicht besonders durchdacht war, verschwand es bald wieder. Die neue MaskedTextBox hat mit ihrer Vorgängerin zum Glück nicht mehr viel zu tun. Sie wurde nicht nur funktional erweitert, sondern sie arbeitet tadellos und weist keine Ungeheimheiten auf. Über die *Mask*-Eigenschaft wird eine Eingabemaske ausgewählt oder festgelegt (es wird bereits eine Reihe von Masken angeboten – die einfachste Regel lautet, dass jede Ziffer durch eine »0« repräsentiert wird). Natürlich lässt sich die Box, die sich wie die *TextBox*-Klasse von der *TextBoxBase*-Klasse ableitet, weiter konfigurieren – die wichtigsten Mitglieder der *MaskedTextBox*-Klasse zeigt Tabelle 9.9. Neben den Eigenschaften spielen auch zwei Events eine Rolle: *MaskChanged* und *MaskInputRejected*. Letzteres wird immer dann ausgelöst, wenn eine Eingabe aus irgendeinem Grund zurückgewiesen wurde.




Abbildung 9.15:
Das MaskedTextBox-Control in Aktion – es stehen bereits ein paar vordefinierte Masken zur Auswahl.



Sie finden das Beispiel auf der Buch-CD in der Projektdatei 09_MaskedTextBox.sln.

Die Validierung der Eingabe

Eine Maske alleine garantiert noch keine valide (sprich gültige) Eingabe. Ein Betrag kann außerhalb eines erlaubten Bereichs liegen oder eine Telefonnummer in der Form nicht existieren. Ganz ohne Validierung geht es daher nicht. Aber es wird gewährleistet, dass der zu überprüfende Wert grundsätzlich in einem bestimmten Format vorliegt. Sowohl eine `TextBox` als auch eine `MaskedTextBox` bieten dafür die Ereignisse *Validating* (Eingabe wird validiert) und *Validated* (Eingabe wurde validiert), die beide von der *Control*-Klasse abstammen. Während beim *Validated*-Event »alles zu spät« ist, kann der *Validating*-Event durch Setzen der *Cancel*-Eigenschaft des *e*-Parameters auf *True*, der vom Typ *CancelEventArgs* ist, abgebrochen werden. Der Fokus verbleibt in diesem Fall in der `TextBox`.

Bei der `MaskedTextBox` gibt es zusätzlich den *TypeValidationCompleted*-Event. Er wird ausgelöst, wenn die Typenvalidierung unter Verwendung jenes Typs, der zuvor der *ValidatingType*-Eigenschaft zugewiesen wurde, abgeschlossen wurde. Damit kann erreicht werden, dass der Inhalt der `MaskedTextBox` einen bestimmten Typ aufweist, zum Beispiel *Short*.

Besitzt die *ValidatingType*-Eigenschaft einen Wert, sieht die Abfolge bei der Validierung bei der `MaskedTextBox` wie folgt aus:

1. Die `MaskedTextBox` verliert den Eingabefokus.
2. Die *ValidateText*-Methode wird mit dem Inhalt der *Text*-Eigenschaft aufgerufen.
3. Die durch die *ValidatingType*-Eigenschaft indirekt festgelegte *Parse*-Methode des Typs (zum Beispiel *Integer*) wird aufgerufen, um aus dem Text einen Wert dieses Typs abzuleiten.
4. Unabhängig davon, ob *Parse* den Inhalt erfolgreich in den Typ konvertieren konnte oder nicht, wird der *TypeValidationCompleted*-Event ausgelöst. Die *IsValid*-Eigenschaft des *e*-Parameters gibt an, ob die Umwandlung erfolgreich war.
5. Es wird der *Validating*-Event aufgerufen, der gegebenenfalls durch Setzen von *Cancel* auf *True* abgebrochen wird.
6. Daraufhin wird der *Validated*-Event ausgelöst. Damit wurde die Eingabe vollständig validiert.
7. Es wird der *LostFocus*-Event aufgerufen.

Die Rolle der `ErrorProvider`

An dieser Stelle wird es Zeit, zumindest mit einem Abstecher auf die *ErrorProvider*-Komponente der Toolbox einzugehen, die sehr nützlich ist, wenn es darum geht, Fehler, die aus einer Eingabevalidierung entstehen, anzuzeigen. Das Konzept der »Windows Forms Provider« ist auf den ersten Blick ein wenig gewöhnungsbe-

dürftig, aber sehr einfach. Der Name rührt von dem Umstand, dass sie anderen Steuerelementen etwas zur Verfügung stellen, in diesem Fall das Anzeigen eines Fehlerzustandes. Wenn Sie einen `ErrorProvider` auf einem Formular anordnen, passiert noch nicht viel. In Aktion tritt er dann, wenn bei einem Steuerelement ein Fehlerzustand angezeigt werden soll, was zum Beispiel im Rahmen der *Validating*-Ereignisprozedur einer Textbox geschieht.

Das folgende Beispiel »schlägt Alarm«, wenn die Eingabe in der Textbox kleiner als 1 und größer als 49 ist:

```
' Klammern sind wichtig, sonst Fehler
If IsNumeric(tbEingabel.Text) AndAlso (tbEingabel.Text < 1 Or
    tbEingabel.Text > 49) Then
    ErrorProvider1.SetError(tbEingabel, "Zahl muss zwischen 1 und 49 liegen")
ElseIf IsNumeric(tbEingabel.Text) = False Then
    ErrorProvider1.SetError(tbEingabel, "Eingabe ist keine Zahl")
Else
    ErrorProvider1.SetError(tbEingabel, "")
End If
```

Dem `ErrorProvider` werden zwei Dinge mitgeteilt: der Namen des Steuerelements und der Text der Fehlermeldung. Anschließend erscheint ein blinkendes rotes Ausrufezeichen rechts neben dem Steuerelement. Aufgehoben wird der Fehlerzustand durch einen erneuten Aufruf von *SetError*, dieses Mal mit einem Leerstring.

Sie finden das Beispiel in der Projektdatei `09_ErrorProvider.sln`.




Abbildung 9.16: Der `ErrorProvider` bietet eine elegante Methode, um Fehler anzuzeigen.

Der Aufbau einer Maske

Der Aufbau einer Eingabemaske ist ausführlich in der Hilfe beschrieben, sodass es hier bei zwei Angaben bleiben soll, die für den Start vollkommen ausreichen: Jede Ziffer wird durch eine »0« repräsentiert. Soll in der Maske kein Promptzeichen, sondern eine 0 erscheinen, muss dieser das Escapezeichen »\« vorausgehen, das generell die Wirkung des folgenden Promptzeichens aufhebt.

Tabelle 9.9:
Interessante Mit-
glieder der Masked-
TextBox-Klasse

Mitglied	Bedeutung
<i>AllowPromptAsInput</i> -Eigenschaft	<i>True</i> , wenn auch das Promptzeichen – normalerweise der Unterstrich (»_«) – als Eingabe erlaubt ist.
<i>AsciiOnly</i> -Eigenschaft	<i>True</i> , wenn nur ASCII-Zeichen im Zeichencodebereich 32 bis 127 erlaubt sind.
<i>BeepOnError</i> -Eigenschaft	<i>True</i> , wenn ein zurückgewiesener (da gemäß der Maskendefinition unzulässiger) Tastendruck mit einem Signalton quittiert werden soll.
<i>CutCopyMaskFormat</i> -Eigenschaft	Eigenschaft vom Enumerationstyp <i>MaskFormat</i> , die festlegt, wie Promptzeichen bei Cut & Copy-Operationen behandelt werden.
<i>HidePromptOnLeave</i> -Eigenschaft	<i>True</i> , wenn die Promptzeichen unsichtbar werden sollen, wenn das Eingabefeld den Fokus verliert.
<i>HideSelection</i> -Eigenschaft	<i>True</i> , wenn der markierte Text nicht hervorgehoben bleiben soll, wenn das Steuerelement den Fokus verloren hat.
<i>InsertKeyMode</i> -Eigenschaft	Legt den Einfügemodus als Konstantenaufflistung vom Typ <i>InsertKeyMode</i> fest.
<i>Mask</i> -Eigenschaft	Legt die Eingabemaske als Zeichenfolge fest (die Hilfe enthält eine Tabelle der erlaubten Zeichen).
<i>RejectInputOnFirstFailure</i> -Eigenschaft	<i>True</i> , wenn die Eingabe nach Auftreten des ersten Eingabefehlers nicht weiter untersucht wird.
<i>ResetOnPrompt</i> -Eigenschaft	<i>True</i> , wenn die Eingabe eines Promptzeichens dazu führt, dass das Promptzeichen wiederhergestellt wird (Default). <i>False</i> , wenn das Promptzeichen als normale Eingabe behandelt wird und der Cursor auf die nächste Position springt.
<i>ResetOnSpace</i> -Eigenschaft	<i>True</i> , wenn die Eingabe eines Leerzeichens dazu führt, dass das Promptzeichen wiederhergestellt wird.
<i>ShortcutsEnabled</i> -Eigenschaft	<i>True</i> , wenn für die TextBox bestimmte Shortcuts erlaubt sein sollen, zum Beispiel <code>[Strg] + [Z]</code> (allgemeine <i>TextBoxBase</i> -Eigenschaft).
<i>SkipLiterals</i> -Eigenschaft	<i>True</i> , wenn die Eingabe von Maskenzeichen (»Literals«) dazu führt, dass die nächste Eingabeposition angesteuert wird. Bei <i>False</i> ist ein Eingabefehler die Folge.
<i>TextMaskFormat</i> -Eigenschaft	Legt als Auflistungskonstante vom Typ <i>MaskFormat</i> fest, ob die Maske auch Prompt- und Literalzeichen enthalten darf.

TextBox mit Formatierung – die RichTextBox

Das Rich Text Format (kurz RTF) galt zwar schon immer als ziemlich »uncool« und als eine Art »HTML für Arme«, doch es besitzt einen unschätzbaren Vorteil. Es funktioniert und tut genau das, was vom ihm erwartet wird: Texte mit Formatierungsmerkmalen anzureichern (und wenn es sein muss, auch mit eingebetteten Bildern), und das in einem Format, das zum Beispiel von jedem gängigen Textprogramm dargestellt werden kann. Lediglich in einem Webbrowser kann es nicht angezeigt werden. Benötigen Sie daher eine TextBox, die auch eine formatierte Eingabe ermöglicht, ist die *RichTextBox*-Klasse der ideale Kandidat. Da das RTF lediglich eine Option ist, eignet sich die *RichTextBox* auch als Alternative zur regulären *TextBox*.

Sie finden das Beispiel auf der Buch-CD in der Projektdatei 09_RTFBox.sln.



Das Geheimnis der Formatierung besteht darin, diese auf den selektierten Text anzuwenden. Netterweise stellt die *RichTextBox*-Klasse mit Eigenschaften wie *SelectionFont* oder *SelectionColor* gleich die passenden Eigenschaften bereit, sodass die Formatierung sehr einfach ist:

```
RichTextBox1.SelectionFont = New Font(RichTextBox1.SelectionFont, _  
    FontStyle.Bold)
```

Wenn Sie sich wundern sollten, dass der Text scheinbar nicht fett formatiert dargestellt wird, liegt dies in der Regel lediglich daran, dass er bereits in Fettschrift dargestellt wird und ein vermeintliches Setzen der *Bold*-Eigenschaft dann keine Auswirkung mehr hat. Man muss daher vorher abfragen, ob die *Bold*-Eigenschaft den Wert *True* hat, und dann das *Bold*-Attribut gegebenenfalls auch zurücksetzen. In der *FontStyle*-Enumeration werden die Attribute *Bold*, *Italic*, *Regular*, *Strikeout* und *Underline* zusammengefasst, die per *Or*-, *And*- oder *Xor*-Operator entsprechend kombiniert werden müssen. Der folgende Befehl formatiert den Text fett, ohne die übrigen Attribute zu beeinflussen:

```
RichTextBox1.SelectionFont = New Font(RichTextBox1.SelectionFont, _  
    RichTextBox1.SelectionFont.Style Or FontStyle.Bold)
```

Der folgende Befehl entfernt die Fettschrift wieder, ohne die anderen Attribute zu verändern:

```
RichTextBox1.SelectionFont = New Font(RichTextBox1.SelectionFont, _  
    RichTextBox1.SelectionFont.Style Xor FontStyle.Bold)
```

Der kleine Unterschied ist ein »X«, genauer gesagt der *Xor*-Operator, der anstelle von *Or* zum Einsatz kommt.

Abbildung 9.17:
Ein formatierter Text
einmal in der
RichTextBox ...



Abbildung 9.18:
... und einmal in
einer bekannten
Textverarbeitung



Tabelle 9.10:
Interessante
Mitglieder der
RichTextBox-Klasse

Mitglied	Bedeutung
<i>AcceptTabs</i> -Eigenschaft	<i>True</i> , wenn das Drücken der <code>Tab</code> -Taste nicht dazu führen soll, dass das nächste Steuerelement angesprungen wird.
<i>AppendText</i> -Methode	Hängt einen Textblock an das Ende der RichText-Box an.
<i>AutoWordSelection</i> -Eigenschaft	<i>True</i> , wenn ein Wort durch einen Doppelklick markiert wird (einen Unterschied zur <i>False</i> -Einstellung konnte der Autor allerdings nicht feststellen).
<i>CanUndo</i> -Eigenschaft	<i>True</i> , wenn die letzte Operation rückgängig gemacht werden kann.
<i>Clear</i> -Methode	Löscht den Inhalt der RichText-Box.
<i>DeselectAll</i> -Methode	Hebt die Auswahl wieder auf.

Mitglied	Bedeutung
<i>DetectUrls</i> -Eigenschaft	<i>True</i> , wenn eine URL speziell formatiert werden soll.
<i>DrawToBitmap</i> -Methode	Rendert den Inhalt der RichTextBox in eine externe Bitmap. Diese mit dem .NET Framework 2.0 eingeführte Methode erbt die Klasse von der allgemeinen <i>Control</i> -Klasse. Beim Rendern werden aber (offenbar) in der Innenfläche angeordnete Steuerelemente und nicht der Text berücksichtigt.
<i>GetCharFormPosition</i> -Methode	Liefert das Zeichen als <i>Char</i> -Wert, das dem übergebenen Punkt am nächsten ist.
<i>GetFirstCharIndexFromLine</i> -Methode	Liefert die Position des ersten Zeichens in der angegebenen Zeile. Auf diese Weise lässt sich eine Zeile heraustrennen.
<i>LoadFile</i> -Methode	Lädt eine Textdatei im RTF- oder ASCII-Format.
<i>RightMargin</i> -Eigenschaft	Liefert oder setzt den rechten Rand des Eingabefeldes.
<i>Rtf</i> -Eigenschaft	Der komplette Inhalt der RichTextBox im RTF-Format.
<i>SaveFile</i> -Methode	Speichert den Inhalt der RichTextBox in eine Datei oder einen Stream.
<i>SelectedText</i> -Eigenschaft	Der markierte Text.
<i>Text</i> -Eigenschaft	Der komplette Inhalt der RichTextBox im ASCII-Format.
<i>Undo</i> -Methode	Macht die letzte Änderung rückgängig.

Tabelle 9.10:
Interessante
Mitglieder der
RichTextBox-Klasse
(Forts.)

HTML-Dokumente anzeigen – das WebBrowser-Steuerelement

Ein Webbrowser ist im Jahre 2006 wahrlich nichts Sensationelles mehr. Auch das Platzieren eines Webbrowsers auf einem Formular in Gestalt eines ActiveX-Controls, sodass dort eine Webseite angezeigt werden konnte, ließ sich im Prinzip bereits bei Visual Basic 4.0 realisieren. Das neue WebBrowser-Control ist dennoch etwas Besonderes, denn es handelt sich nicht mehr um ein ActiveX-Control, das lediglich in einem »Korsett« auf ein Windows-Formular gehievt wird. Die Browser-Engine wurde vielmehr in eine Klasse eingehüllt, die sich von der *Control*-Klasse ableitet und damit nicht nur einen konsistenten Satz an Eigenschaften und Methoden anbietet, sondern auch den üblichen Komfort eines Steuerelements aufweist wie eine *DrawToBitmap*-Methode, mit der sich der Inhalt des angezeigten HTML-Dokuments in eine Bitmap umwandeln lässt.



Das folgende Beispiel lädt eine Webseite in das WebBrowser-Control:

```
Dim Ur1 As New Uri("http://www.mut.de")
wb.Navigate(Ur1)
```

Abbildung 9.19:
Das WebBrowser-
Control zeigt den
Inhalt der geladenen
Webseite an.



Zugriff auf das DOM

Ist ein HTML-Dokument erst einmal im Kasten, sprich im Arbeitsspeicher des Programms, möchte man vielleicht auch auf dessen Inhalt zugreifen. Hier kommt das *Document Object Model* (DOM) ins Spiel, das jedes HTML-Dokument als Objektbaum repräsentiert, dessen Objekte mit den Seitenelementen korrespondieren und dessen Struktur der stets hierarchischen Struktur einer HTML-Seite entspricht. Damit lassen sich zum Beispiel alle Tags, alle Frames, alle Skriptblöcke und deren Inhalte nicht nur ansprechen und enumerieren, sondern auch verändern.



Der folgende Codeschnipsel listet im DocumentCompleted-Ereignis die Namen aller Tags des geladenen Dokuments auf:

```
Dim Wb As WebBrowser = CType(sender, WebBrowser)
1bDOM.Items.Clear()
For Each T As HtmlElement In Wb.Document.All
    If T.TagName <> "" Then 1bDOM.Items.Add(T.TagName)
Next
```

Die Möglichkeit, HTML-Dokumente auch editieren zu können, bietet das Web-Browser-Control nicht. Wer das möchte, muss nach wie vor auf die (inzwischen recht betagte) COM-Komponente DHTML Editor Control zurückgreifen, die Bestandteil des Internet Explorers ist und daher auf jedem System vorhanden sein sollte (einen anschaulichen Artikel, der dieses im Grunde recht interessante Control in Aktion zeigt, finden Sie unter <http://www.microsoft.com/germany/msdn/library/web/DasDHTMLEditorControlMitVBNETVerwendet.mspix>).



Den Webbrowserinhalt als Bitmap speichern

Eine einfache »Screenshot«-Funktion lässt sich durch den Umstand realisieren, dass sich auch die *WebBrowser*-Klasse als echte Steuerelementklasse von der *Control*-Klasse ableitet und diese seit dem .NET Framework 2.0 eine *DrawToBitmap*-Methode anbietet, mit der sich der Inhalt eines Steuerelements in eine Bitmap »rendern« lässt (so wird das Umwandeln in etwas Sichtbares allgemein bezeichnet).

Das folgende Beispiel »schießt« ein Bildschirmfoto von dem Inhalt des *WebBrowser*-Control-Fensters, speichert dieses als Bitmap und zeigt es anschließend über die *Start*-Methode der *Process*-Klasse mit dem Programm *Paint* an:



```
Dim Bm As New Bitmap(wb.ClientRectangle.Width, _
    wb.ClientRectangle.Height)
wb.DrawToBitmap(Bm, wb.ClientRectangle)
' Nicht zwingend erforderlich
wb.BringToFront()
Bm.Save("C:\WbBitmap.bmp")
Process.Start("MsPaint", "C:\WbBitmap.bmp")
```

Anzeigen von animierten Gifs

Im Zusammenhang mit dem letzten Beispiel kommt ein weiterer Aspekt ins Spiel, der sich unter VB6 nicht ohne Hilfsmittel lösen ließ: das Anzeigen und vor allem Steuern von animierten Gif-Dateien. Die Aufgabenstellung könnte etwas folgendermaßen aussehen: Während ein HTML-Dokument geladen wird, soll ein rotierender Globus dargestellt werden, der aus einer simplen animierten Gif-Datei (*World.gif*) mit einer Reihe von Bitmaps besteht, die nacheinander erscheinen. Das Anzeigen eines animierten Gifs ist grundsätzlich kein Problem. Es wird einfach ein *Bitmap*-Objekt instanziiert und der *Image*-Eigenschaft einer *PictureBox* zugewiesen:

```
Dim Bm As New Bitmap("World.gif")
pbStatus.Image = Bm
```

Um das animierte Gif auch steuern zu können, bedarf es der *ImageAnimator*-Klasse und ihrer freigegebenen Methoden *Animate* und *StopAnimate*:

```
ImageAnimator.Animate(pbStatus.Image, FrameEv)
```

In beiden Fällen muss ein simpler Eventhandler übergeben werden, der eine Ereignisprozedur definiert, die aufgerufen wird, wenn die Animation läuft:

```
Private FrameEv As New EventHandler(AddressOf FrameChange)
```

Auch wenn die Eventprozedur (pro Frame) aufgerufen wird, passiert hier nicht viel, zumal aufgrund des allgemeinen Typs des Eventhandlers zum Beispiel nicht die Framenummer übergeben wird.

Tabelle 9.11:
Interessante Mitglieder der Web-Browser-Klasse

Mitglied	Bedeutung
<i>AllowNavigation</i> -Eigenschaft	<i>True</i> , wenn das Navigieren innerhalb der bereits angezeigten Seiten erlaubt sein soll.
<i>Document</i> -Eigenschaft	Steht für das HTML-Objektmodell, das jedes Seitenelement durch ein entsprechendes Objekt repräsentiert.
<i>GoBack</i> -Methode	Ruft die zuletzt angezeigte Seite auf.
<i>Print</i> -Methode	Druckt den Seiteninhalt aus.
<i>ScriptErrorsSuppressed</i> -Eigenschaft	<i>True</i> , wenn Skriptfehler beim Laden eines Dokuments nicht angezeigt werden sollen (die Standardeinstellung ist <i>False</i>).
<i>StatusText</i> -Eigenschaft	Steht für den aktuellen Zustand des Web-Browsers als Zeichenkette.

9.4 Ziehen-und-Ablegen-Operationen

Bei einer Ziehen-und-Ablegen-Operation wird ein Steuerelement bei gedrückter Maustaste an einen anderen Ort innerhalb desselben oder eines anderen Formulars gezogen und dort abgelegt. Sie sind bei WinForms-Formularen und ihren Steuerelementen fest eingebaut. Gegenüber der »neumodischen« Variante des Ziehens und Ablegens bei VB6 gibt es nur weniger Unterschiede.



Die Drag & Drop-Implementierung bei den Windows Forms-Formularen ist dem OLE-Drag & Drop früherer Visual Basic-Versionen sehr ähnlich.

Ziehen und Ablegen starten

Eine Ziehen-und-Ablegen-Operation ist mit jedem Steuerelement möglich, sie muss aber explizit gestartet werden. Dies geschieht über die *DoDragDrop*-Methode des Steuerelements, die am besten im *MouseDown*-Ereignis aufgerufen wird.



Das folgende Beispiel startet eine Ziehen-und-Ablegen-Operation mit der ListBox liSpieler, wenn die linke Maustaste gedrückt wird:

```
If e.Button = Windows.Forms.MouseButtons.Left Then
    liSpieler.DoDragDrop(liSpieler.SelectedItem, DragDropEffects.Copy)
End If
```

Der *DoDragDrop*-Methode werden zwei Argumente übergeben:

- ➔ das zu ziehende Element (in diesem Fall der aktuell selektierte Listeneintrag, bei dem es sich auch um ein Objekt handeln kann),
- ➔ die durchzuführende Operation, die über die Enumeration *DragDropEffects* ausgewählt wird.

Im obigen Fall wurde *Copy* gewählt. Dies betrifft in erster Linie das visuelle Feedback (je nachdem, ob *Copy* oder *Move* verwendet wird, wird ein anderes Symbol angezeigt), hat aber keinen Einfluss auf die tatsächlich durchgeführte Operation, denn diese wird vom Programmierer festgelegt. Die Auswahl dient also lediglich dazu, den (optischen) Modus der Operation festzulegen, sodass der Mauszeiger die entsprechende Form erhält. Was im Einzelnen passiert, spielt keine Rolle.

Auf eine Ablegen-Operation reagieren

Damit ein Steuerelement der Empfänger einer Ablegen-Operation werden kann, muss dessen *AllowDrop*-Eigenschaft auf *True* gesetzt werden. Ist dies geschehen, wird eine Reihe von Ereignissen ausgelöst, sobald sich ein Objekt im »Hoheitsgebiet« (also in der sichtbaren Fläche) des Steuerelements befindet (siehe Tabelle 9.12). Die wichtigsten Ereignisse sind *DragEnter* und *DragDrop*. Bei *DragEnter* betritt das Steuerelement den Bereich des potenziellen Ziels, und dieses muss gegebenenfalls ein visuelles Feedback geben, indem es das Symbol des Mauszeigers ändert. Bei *DragDrop* wird das Steuerelement abgelegt, und das Zielobjekt muss darauf nun irgendwie reagieren.

Bei beiden Ereignissen ist der *e*-Parameter vom Typ *System.Windows.Forms.DragEventArgs*. Dies ist der Schlüssel zu dem, was abgelegt wurde (siehe Tabelle 9.13). Es kann eine Bitmap, ein Text oder ein beliebiges Objekt sein. Das macht das Ziehen und Ablegen sehr leistungsfähig. An die Daten kommt man stets über die *GetData*-Methode heran.

Das folgende Beispiel ist der Inhalt einer DragEnter-Ereignisprozedur. Sie prüft, ob es in einem Panel paDreamTeam bereits ein Label gibt, dessen Text-Eigenschaft dem Wert der Eigenschaft SpielerName des gezogenen Spieler-Objekts entspricht. Sollte dies der Fall sein, wird die Aktion abgebrochen und ein späteres Ablegen verhindert:

```
If e.AllowedEffect = DragDropEffects.Copy Then
    Dim Sp As Spieler = e.Data.GetData(GetType(Spieler))
    ' Feststellen, ob der Spieler bereits gedroppt wurde
    For Each C As Control In paDreamTeam.Controls
        If TypeOf C Is Label Then
            If Sp.Spielername = CType(C, Label).Text Then
                e.Effect = DragDropEffects.None
                Exit Sub
            End If
        End If
    End For
    Next
    e.Effect = DragDropEffects.Copy
End If
```



Diese Prozedur enthält zwei wichtige Programmier-Techniken:

- ➔ Über *GetData* wird gezielt auf den Typ *Spieler* zugegriffen, der durch *DoDragDrop* übergeben wurde. Der Typ, also das *Type*-Objekt, das hinter der Klasse *Spieler* steht (alles Weitere im nächsten Beispielprojekt), wird über die *GetType*-Methode abgefragt.
- ➔ In einer *For Each*-Schleife werden alle Labels durchlaufen und ihre *Text*-Eigenschaft geprüft. Bei einer Übereinstimmung wird durch Setzen der *Effect*-Eigenschaft erreicht, dass ein »Ablegen nicht möglich«-Symbol eingeblendet wird. Der Anwender erkennt daran, dass er das Objekt (aus welchen Gründen auch immer) nicht auf dem Panel ablegen kann. Ist der Name dagegen noch nicht vorhanden, wird das *Copy*-Symbol beibehalten, und der geschulte Anwender erkennt, dass eine Kopie des Objekts abgelegt wird.

Tabelle 9.12:
Die Ziehen- und
Ablegen-Ereignisse
in der Übersicht

Ereignis	Auslösende Aktion
<i>DragDrop</i>	Das gezogene Element wird auf dem Steuerelement abgelegt.
<i>DragEnter</i>	Das gezogene Element betritt den sichtbaren Bereich des Steuerelements.
<i>DragLeave</i>	Das gezogene Element verlässt den sichtbaren Bereich des Steuerelements wieder.
<i>DragOver</i>	Das gezogene Element bewegt sich im Bereich des Steuerelements.

Tabelle 9.13:
Die Elemente der
DragEventArgs-
Klasse

Element	Bedeutung
<i>AllowedEffect</i> -Eigenschaft	Nur-lesen-Eigenschaft, die alle für das Objekt erlaubten und beim Aufruf von <i>DoDragDrop</i> festgelegten Effekte enthält.
<i>Data</i> -Eigenschaft	Steht für ein <i>DataObject</i> -Objekt, das einen Zugriff auf die übergebenen Daten erlaubt. Dieses Objekt umfasst die Methoden <i>GetData</i> , <i>GetDataPresent</i> , <i>GetFormats</i> und <i>SetData</i> .
<i>Effect</i> -Eigenschaft	Legt fest, welches Symbol der Mauszeiger im weiteren Verlauf der Operation annehmen soll.
<i>KeyState</i> -Eigenschaft	Steht für den aktuellen Zustand der Tasten  ,  und  sowie der Maustasten.
X-Eigenschaft	x-Position des Mauszeigers.
Y-Eigenschaft	y-Position des Mauszeigers.



Sie finden das Beispiel auf der Buch-CD in der Projektdatei 09_PanelDragDrop.sln.

Das folgende Beispiel stellt ein vollständiges Beispiel für eine Ziehen-und-Ablegen-Operation dar. Es besteht aus einem WinForms-Formular, auf dem sich eine ListBox und ein Panel als Spielplan befinden: *liSpieler* und *paDreamTeam*. Fußballfans wissen sicherlich, worum es geht. Für alle anderen: Die Liste enthält eine (willkürliche) Auswahl von Topspielernamen, die auf das Panel mit der Maus gezogen und dort nicht nur abgelegt, sondern auch verschoben werden können (warum nicht, denn das Drag & Drop funktioniert auch innerhalb eines Steuerelements). Da dabei ein *Spieler*-Objekt gezogen wird, kann die Bitmap des Landes, für das der Spieler antritt, angezeigt werden (besser wäre natürlich ein Foto des Spielers, doch dann gibt es unter Umständen Ärger mit den Sponsoren²).

Das Listing wird im Folgenden nicht komplett abgebildet, sondern nur die für die Ziehen-und-Ablegen-Operationen wichtigen Bereiche.

Der entscheidende Part ist, dass sich auf einem Panel insgesamt elf *PictureBox*-Controls befinden. In jedem soll das Ablegen eines *Spieler*-Objekts möglich sein. Bei VB6 hätte man diese PictureBoxen zu einem Steuerelementfeld zusammenfassen müssen. Bei Visual Basic 2005 kann dies nachträglich geschehen:

```
' Alle PictureBoxen des Panels zu einem Feld zusammenfassen
For Each C As Control In paDreamTeam.Controls
  If TypeOf C Is PictureBox Then
    PictureFeld(Nr) = C
    With CType(C, PictureBox)
      .AllowDrop = True
      .SizeMode = PictureBoxSizeMode.StretchImage
    End With
    Nr += 1
    AddHandler CType(C, PictureBox).DragDrop, _
      AddressOf Picbox_DragDrop
    AddHandler CType(C, PictureBox).DragEnter, _
      AddressOf Picbox_DragEnter
  End If
Next
```

Alle PictureBoxen teilen sich denselben *DragDrop*-Handler, in dem Folgendes passiert:

- ➔ Der *Image*-Eigenschaft wird der Wert der *Flagge*-Eigenschaft des *Spieler*-Objekts zugewiesen, die für eine zuvor zugewiesene Flaggen-Bitmap steht.
- ➔ Über den Namen der PictureBox wird das dazugehörige Label lokalisiert und dort der Name des Spielers eingetragen (es wird davon ausgegangen, dass, wenn eine PictureBox zum Beispiel »PictureBox1« heißt, das dazugehörige Label den Namen »Label1« trägt).

² Das ist natürlich ein wenig ironisch gemeint, wobei man bei der regelwütigen FIFA nie ganz sicher sein kann. Aber generell muss man die Persönlichkeitsrechte bei prominenten Menschen beachten.


```

Sub Picbox_DragDrop(ByVal sender As Object, ByVal e As System. _
    Windows.Forms.DragEventArgs)
    If e.Effect = DragDropEffects.Copy Then
        Dim Sp As Spieler = e.Data.GetData(GetType(Spieler))
        CType(sender, PictureBox).Image = Sp.Flagge
        Dim PicName As String = CType(sender, PictureBox).Name
        ' Nr extrahieren
        Dim Nr As String
        For i As Short = 0 To PicName.Length - 1
            Dim C As Char = PicName.Chars(i)
            If Char.IsNumber(C) Then
                Nr = PicName.Substring(i)
            End If
        Next
        CType(paDreamTeam.Controls.Find("Label" & Nr, False)(0), Label). _
            Text = Sp.Spielername
    End If
End Sub

```

Abbildung 9.20:

Wer ist das Dreamteam 2006? Dank Drag & Drop können Sie sich einfach und elegant eine eigene Auswahl zusammenstellen.



9.5 Layout ist mehr als nur das Anordnen von Steuerelementen

Hinter einer Benutzeroberfläche, die alle wichtigen Anforderungen bezüglich Ergonomie, Bedienbarkeit und vor allem Konsistenz erfüllt, steckt sehr viel Arbeit. Arbeit, die der normale Anwender in der Regel nicht sieht, aber als selbstverständlich voraussetzt³. Auch Visual Basic 2005 zaubert keine perfekte Oberfläche auf Knopfdruck, bietet aber ein paar Hilfestellungen, die es bei früheren Versionen nicht gab:

- ➔ Die Führungslinien, die beim Anordnen eines Steuerelements sicherstellen, dass es bündig zu anderen Steuerelementen angeordnet wird. Diese Neuerung ist einfach und genial.
- ➔ Einen »Griff«, der beim Selektieren eines Steuerelements in Gestalt eines kleinen Vierfachpfeils sichtbar wird und über den sich das Steuerelement verschieben lässt (Sie müssen es also nicht mehr irgendwo zu »fassen« kriegen).
- ➔ Ein `TableLayout`-Control, mit dem sich die Steuerelemente anhand einer unsichtbaren Tabelle wie auf einer HTML-Seite anordnen lassen.
- ➔ Eine Undo-Funktion, die auch das Verschieben von Steuerelementen rückgängig macht (probieren Sie es einmal aus).
- ➔ Verschiedene Verbesserungen, die dafür sorgen, dass sich der Inhalt eines Formulars variabel an die Größe des Formulars anpasst.
- ➔ Eine neue Layout-Engine, die für eine modernere Optik sorgt (das typische, altmodische »Visual Basic Look & Feel«, an dem man früher treffsicher eine Visual Basic-Anwendung erkennen konnte, gibt es nicht mehr).
- ➔ Durch den Umstand, dass ein Formular vollständig auf Quelltext basiert, ist es für ein Add-In problemlos möglich, ein Formular nachzubearbeiten, indem zum Beispiel die Größe oder auch die Position eines Steuerelements anhand einer Vorgabe überprüft und angepasst wird.

Trotz Führungslinien ist das Anordnen von Steuerelementen oft noch ein Geduldspiel. Am einfachsten ist es daher, die Position und Größe in den zuständigen Eigenschaften *Location* und *Size* im Eigenschaftenfenster auf den Pixel genau einzustellen.

Steuerelemente verankert – die *Anchor*-Eigenschaft


Über die *Anchor*-Eigenschaft, die jedes Steuerelement besitzt, wird es möglich, dieses mit einer oder mehreren der vier umgebenden »Seitenwände« zu verankern, sodass es diesen Abstand beim Vergrößern und Verkleinern des Formulars beibehält, was wiederum dazu führt, dass es seine Größe anpasst⁴. Verankern bedeutet aber nicht andocken, das heißt, das Steuerelement ändert seine Position nur relativ.

Die *Anchor*-Eigenschaft, die am einfachsten über das Eigenschaftenfenster eingestellt wird, kann insgesamt fünf Einstellungen annehmen: *Bottom*, *Left*, *None*, *Right* und *Top*. Da ein Steuerelement mit mehreren Rändern verankert sein kann, werden die Einstellungen per *Or*-Operator verknüpft.

3 Die Erwartungen sind auch in diesem Bereich grenzenlos. Und wer muss sie im Allgemeinen erfüllen? Der »arme« Entwickler.

4 Dies ist praktisch der lang erwartete Ersatz (es gibt ihn bereits seit Visual Studio .NET 2002) für jene »Resizer-Controls«, die bereits zu Zeiten von Visual Basic 3 populär waren.

Abbildung 9.21:
Die Anchor-Eigen-
schaft sorgt dafür,
dass der relative
Abstand zu den
ausgewählten
Rändern konstant
bleibt.



Steuerelemente andocken – die Docking-Eigenschaft

Das Andocken eines Steuerelements ist eine Eigenschaft, die bei früheren Visual Basic-Versionen fest eingebaut war. Einige »Spezialelemente« wie die Tool- oder Statusbar besaßen die Fähigkeit, automatisch an die obere oder untere Umrandung zu »hüpfen«. Bei .NET weist jedes Steuerelement eine *Docking*-Eigenschaft auf, durch die es an einem oder allen der vier Seitenränder andocken kann. Ein Steuerelement, das an alle Seiten andockt, füllt die Innenfläche des Formulars nicht nur komplett aus, sondern ändert seine Größe auch entsprechend. Der *SplitContainer* ist ein Beispiel für ein Steuerelement, bei dem dieses Verhalten sehr gut passt. Das Andockverhalten wird über die *Dock*-Eigenschaft eingestellt, die insgesamt sechs Einstellungen kennt: *Bottom*, *Fill* (alle Ränder werden andockt), *Left*, *None*, *Right* und *Top*. Auch hier gilt, dass es am einfachsten ist, die Einstellungen im Eigenschaftenfenster auszuwählen.




Abbildung 9.22: Über die Dock-Eigenschaft wird ein Andocken des Steuerelements an einem oder allen Rändern erreicht.

Automatische Größenanpassung

Eine automatische Größenanpassung, die durch Setzen einer »AutoSize«-Eigenschaft auf *True* erreicht werden könnte, gibt es auch bei Visual Basic 2005 nicht (und damit keinen direkten Ersatz für die Resizer-Controls früherer Versionen). Wem eine solche automatische Anpassung wichtig ist, erreicht diese über verschiedene Maßnahmen:

- ➔ Über das Verankern eines Steuerelements an allen vier Rändern – damit passt sich aber die Schriftgröße nicht automatisch an.
- ➔ Über den *Resize*-Event, der bei der Größenänderung eines Steuerelements eintritt. Da es problemlos möglich ist, sämtliche *Resize*-Events aller Steuerelemente (über eine kleine Schleife, welche die *Controls*-Auflistung rekursiv durchläuft) auf einen Handler zu vereinigen, besteht die Möglichkeit, eine zentrale »Anpassungsroutine« einzurichten.

Die AutoSize-Eigenschaft

Es gibt sie doch bei .NET 2.0 – eine *AutoSize*-Eigenschaft. Allerdings nicht für das gesamte Formular, sondern nur pro Steuerelement und mit der Wirkung, dass sich die Größe des Steuerelements an seinen Inhalt (etwa bei einem Label), aber nicht an die Größe des umgebenden Containers anpasst.

Layout in der Tabelle – das `TableLayoutPanel`-Control

Wer sich bereits ein wenig mit HTML und der Umsetzung kleiner Webanwendungen beschäftigt hat, weiß wahrscheinlich, dass der `<Table>`-Tag und die damit einhergehenden Tabellen eine beliebte, wenn auch bei Layoutpuristen sehr umstrittene Möglichkeit sind, ein »Layout« zu gestalten⁵. Die Idee ist simpel: Eine unsichtbare Tabelle gibt mit ihren Zeilen und Spalten Positionen vor, an denen Seitenelemente positioniert werden. Da eine pixelgenaue Positionierung bei HTML ansonsten sehr viel Erfahrung voraussetzt und bei alten Browsern nicht funktioniert, sind Tabellenlayouts nach wie vor sehr beliebt.

Mit .NET 2.0 bietet das `TableLayoutPanel`-Control einen ähnlichen Komfort, wengleich mit HTML absolut keine Gemeinsamkeiten zu verzeichnen sind. Die Idee ist auch hier, dass eine zur Ausführungszeit unsichtbare Tabelle das Anordnen von Steuerelementen auf einem Formular erleichtern soll. Auch wenn das alles nichts mit HTML und dem Web zu tun hat, liegt es nahe, dass sich ein solches Formular vergleichsweise einfach auf eine HTML- oder besser eine XAML-Oberfläche übertragen lässt. XAML steht für *Extended Application Markup Language* und ist bei Microsoft die Zukunft des Oberflächendesigns für Windows- wie Webanwendungen gleichermaßen. Mit der nächsten Version von Visual Studio wird es einen integrierten XAML-Designer geben, sodass wir alle in ein paar Jahren nur noch mit XAML programmieren. Dann wird auch die automatische Größenanpassung inklusive sein. Aber das ist natürlich ein anderes Thema⁶.

9.6 Visual Styles bei Windows XP und das Windows-Anwendungsframework

Auf eine »saubere« Unterstützung der mit Windows XP eingeführten »Visual Styles« mussten Programmierer etwas länger warten⁷, was insofern schade ist, da es inzwischen Dutzende, wenn nicht Hunderte von Styles gibt. Ob es sinnvoll ist, seinen Computer mit einer solchen »Performancebremse« auszustatten, sei einmal dahingestellt, eine Anwendung sollte aber in der Lage sein, sich automatisch an das aktuelle Design anzupassen. Unter Visual Studio .NET 2003 musste noch vor dem Aufruf von `InitializeComponent` in der Formalklasse der Aufruf von `Application.EnableVisualStyles()` erfolgen. Wurde bei einem Steuerelement, wie einem Button, für die `FlatStyle`-Eigenschaft `System` eingestellt, übernahm ein Steuerelement den vorgegebenen Style (bei den meisten Steuerelementen war es nicht erforderlich). Unter Visual Studio 2005 ist dieser Zusatzschritt nicht mehr erforderlich, die Steuerelemente passen sich (bereits zur Entwurfszeit) automatisch an. Ist dies aus irgendeinem Grund nicht erwünscht, muss lediglich die Option `VISUAL XP-STILE AKTIVIEREN` in den Anwendungseigenschaften ausgeschaltet werden.

5 Solche Spaßverderber gibt es leider überall.

6 Wenn Sie möchten, können Sie sich eine Vorabversion von »Cider«, so der Name dieses Designers, für Visual Studio 2005 bereits heute ansehen. Oder Sie lesen diese Zeilen erst zu einem Zeitpunkt, zu dem Visual Studio 2007 schon offiziell ist, und wundern sich vielleicht, dass es noch etwas anderes als XAML geben könnte.

7 Immerhin gibt es Windows XP bereits seit Oktober 2001.

Hinter den Styles steckt eine altbekannte Komponente: Comctl32.dll (welche die »Windows Common Controls« enthält), die inzwischen in der Version 6 vorliegt (und nur unter Windows XP und Windows Server 2003 läuft). Unter Windows Server 2003 muss gegebenenfalls der Systemdienst Themes aktiviert und gestartet werden.




Abbildung 9.23: Zu den Anwendungseigenschaften gehört auch die Option zum Aktivieren der XP-Stile.

9.7 Es gibt sie doch – die Steuerelementfelder

Dieser Abschnitt richtet sich an jene VB6-Programmierer, die immer noch ein wenig ihren Steuerelementfeldern nachtrauern, die es bei .NET nicht mehr gibt. Sie können innerhalb des Windows Forms-Designers Steuerelemente beliebig kopieren und einfügen, ohne dass dadurch ein Steuerelementfeld entsteht. Das heißt aber nicht, dass Steuerelemente nicht mehr zu einem Feld zusammengefasst werden können. Diese Möglichkeit existiert weiterhin, und sie ist deutlich flexibler als früher. Die entsprechende Funktion ist nur nicht mehr zur Entwurfszeit verfügbar, sondern ausschließlich zur Ausführungszeit.

Da Steuerelemente reguläre Klassen sind, kann jedes Steuerelement zur Laufzeit durch Instanzieren der dazugehörigen Klasse aus dem »Nichts« heraus entstehen:

```
Dim bnNeu As New Button()
```

Damit steht *bnNeu* für einen Button. Wird dieser zur *Controls*-Auflistung eines Containers hinzugefügt, wird der Button tatsächlich angezeigt, wenngleich man in der Praxis Größe, Position und andere Attribute setzen würde:

```
Me.Controls.Add(bnNeu)
```

Benötigt man nicht einen, sondern mehrere Buttons, fasst man die Befehle einfach in einer Schleife zusammen und fügt die resultierenden Instanzen in ein Array oder in eine Auflistung ein:

```
Dim ButtonListe As New List(Of Button)
```

```
For i As Short = 1 To 26
    bnNeu = New Button()
    ButtonListe.Add (bnNeu)
    Me.Controls.Add (bnNeu)
Next
```

Am Ende enthält die Auflistung *ButtonListe* 26 Buttons. Zwei wichtige Anforderungen fehlen aber noch: Die Buttons müssen unterscheidbar sein, und sie müssen auf Ereignisse reagieren können. Den ersten Punkt erreicht man über die universelle *Tag*-Eigenschaft, den zweiten über den *AddHandler*-Befehl:

```
For i As Short = 1 To 26
    bnNeu = New Button()
    bnNeu.Tag = CType(i,String)
    AddHandler bnNeu.Click, AddressOf ButtonSammelClick()
    ButtonListe.Add (bnNeu)
    Me.Controls.Add (bnNeu)
Next
```

ButtonSammelClick ist eine Prozedur, die bei jedem Klick auf einen der Buttons aufgerufen wird. Über die *Tag*-Eigenschaft des Buttons, die über den *Sender*-Parameter übergeben wird, kann festgestellt werden, welcher Button angeklickt wurde:

```
ButtonNr = CType(Sender, Button).Tag
```

Damit steht die Funktionalität eines Steuerelementfeldes auch unter Visual Basic 2005 zur Verfügung, nur auf dem Formular zusammenstellen lässt es sich nicht⁸.



Sie finden das Beispiel auf der Buch-CD in der Projektdatei 09_ButtonFeld.sln.



Das folgende Beispiel ordnet 26 Buttons in einem Panel an, sodass diese die Innenfläche des Panels optimal ausfüllen. In einer Schleife wird bei jedem Durchlauf ein neuer Button angelegt, mit Eigenschaftswerten ausgestattet und am Ende an die Controls-Auflistung des Panels gehängt:

```
Dim B As Button
Dim XPos, YPos As Single
YPos = 10
For i As Short = 1 To 26
    B = New Button
    With B
        .Text = Chr(64 + i)
        If XPos + .Width > Panel1.Width Then
```

⁸ Denkbar wäre es, einen Designer zu programmieren, der eine entsprechende Funktionalität ermöglicht. Der Aufwand würde aber in keinem Verhältnis zum Nutzen stehen.

```

XPos = 10
YPos += 32
End If
.Width = 36
.Height = 24
.Top = YPos
.Left = XPos
XPos += 48
End With
Panel1.Controls.Add(B)
AddHandler B.Click, AddressOf ButtonClick
Next

```

Der gemeinsame Click-Handler macht nichts anderes, als die Text-Eigenschaft des angeklickten Buttons auszugeben:

```

Sub ButtonClick(ByVal Sender As Object, ByVal e As EventArgs)
    MessageBox.Show("Gib mir ein " & Sender.Text)
End Sub

```



Abbildung 9.24:
Im Prinzip gibt es Steuerelementfelder auch bei Visual Basic 2005.

9.8 MDI-Anwendungen

Eine MDI-Anwendung ist eine reguläre Windows-Anwendung, bei der lediglich das Zusammenspiel der Fenster einfachen Regeln gehorcht, die als »Multiple Document Interface« (MDI) bezeichnet werden und schon »uralt« sind. Demnach

- ➔ besitzt eine Anwendung ein Hauptfenster, in dessen Innenfläche sich alle übrigen Fenster anordnen. Das Hauptfenster ist das MDI-Elternfenster, die übrigen Fenster stellen die MDI-Kindfenster dar (eine Anwendung kann aber auch mehrere MDI-Hauptfenster aufweisen).
- ➔ kann jedes Kindfenster nur in der Innenfläche des Hauptfensters verschoben werden. Wird ein Kindfenster auf Maximalgröße vergrößert, füllt es die Innenfläche des Elternfensters vollständig aus. Seine Titelleiste wird in der Titelleiste des Hauptfensters angezeigt.

- ➔ lassen sich alle Kindfenster überlappend, nebeneinander, untereinander und als Symbole verkleinert zusammenstellen.
- ➔ werden alle Kindfenster in einem *Fenster*-Menü aufgelistet.

MDI-Anwendungen sind immer dann sehr praktisch, wenn eine Anwendung mit einer beliebigen Anzahl an Dokumentfenstern arbeitet, die mit möglichst wenig Aufwand verwaltet werden sollen und die sich an bestimmte Regeln anpassen, die bei allen MDI-Anwendungen gleich sind. MDI-Anwendungen sind aber nur eine Option. Sie können das Thema komplett ignorieren, ohne etwas Wichtiges zu verpassen. MDI-Anwendungen besitzen aber nach wie vor ihre Berechtigung, sodass im Folgenden ein kleiner Bildbetrachter als MDI-Anwendung vorgestellt werden soll.



Sie finden das Beispiel auf der Buch-CD in der Projektdatei 09_MdiBildbetrachter.sln.

Ein Formular zum MDI-Hauptfenster machen

MDI-Formulare gibt es nicht mehr, ein Formular wird durch Setzen der *IsMdiContainer*-Eigenschaft auf *True* zum MDI-Hauptfenster.

Ein Formular zum MDI-Kindfenster machen

Eine Eigenschaft, die ein Formular zum MDI-Kindfenster macht, gibt es nicht. Ein Formular wird zum MDI-Kindfenster, indem seine *MdiParent*-Eigenschaft zur Laufzeit mit einem Verweis auf das MDI-Hauptfenster belegt wird.



Die folgende Click-Prozedur eines ToolStrip-Buttons zeigt einen Dateiauswahldialog zum Laden einer Bitmapdatei an und stellt diese anschließend in der Innenfläche eines Formulars dar, das zunächst komplett neu angelegt und dann zum MDI-Kindfenster wird:

```
Using Ofd As New OpenFileDialog
    With Ofd
        .InitialDirectory = Environment.GetFolderPath _
            (Environment.SpecialFolder.MyPictures)
        If .ShowDialog = Windows.Forms.DialogResult.OK Then
            Dim MdiForm As New Form
            With MdiForm
                .BackgroundImage = New Bitmap(Ofd.FileName)
                .BackgroundImageLayout = ImageLayout.Stretch
                .Text = Path.GetFileName(Ofd.FileName)
                .MdiParent = Me
                .Show()
            End With
        End If
    End With
End Using
```

Auf das aktive MDI-Kindfenster zugreifen

Das aktive MDI-Kindfenster steht über die Eigenschaft *ActiveMdiChild* des Formulars zur Verfügung, ein (*Form*-)Array mit allen MDI-Kindfenstern über die Eigenschaft *MdiChildren*.

Das folgende Beispiel invertiert das Hintergrundbild des aktiven MDI-Kindfensters (mithilfe einiger »GDI-Tricks«, genauer gesagt Klassen im Namespace `System.Drawing.Imaging`, um die es in Kapitel 22 gehen wird.



```
If Me.ActiveMdiChild IsNot Nothing Then
    Dim Im As New ImageAttributes
    Dim Img As Image = Me.ActiveMdiChild.BackgroundImage
    Dim Bm As New Bitmap(Img.Width, Img.Height)
    Dim Cm As New ColorMatrix
    With Cm
        .Matrix00 = -1
        .Matrix11 = -1
        .Matrix22 = -1
    End With
    Im.SetColorMatrix(Cm)
    Dim G As Graphics = Graphics.FromImage(Bm)
    G.DrawImage(Img, New Rectangle(0, 0, Img.Width, Img.Height), 0, _
        0, Img.Width, Img.Height, GraphicsUnit.Pixel, Im)
    G.Dispose()
    Img.Dispose()
    Me.ActiveMdiChild.BackgroundImage = Bm
End If
```

Das Fenster-Menü implementieren

Das typische *Fenster*-Menü einer MDI-Anwendung, das die Namen aller MDI-Kindfenster enthält, lässt sich sehr einfach realisieren, wenngleich es beim neuen *MenuStrip*-Control gegenüber dem *MainMenu*-Control des .NET Frameworks 1.1 eine kleine Änderung gegeben hat: Damit ein Menü, also ein *MenuStrip*-Control, die Fensterliste anzeigt, muss die *MdiWindowListItem*-Eigenschaft des *MenuStrip*-Controls mit der Referenz auf jenes *ToolStripMenuItem*-Objekt gelegt werden, das die Fenster anzeigen soll. Heißt dieses zum Beispiel *mnFenster*, lautet der Befehl wie folgt:

```
MenuStrip1.MdiWindowListItem = mnFenster
```

Über die (mit .NET 2.0 eingeführte) *IsMdiWindowListEntry*-Eigenschaft lässt sich feststellen, ob ein *ToolStripMenuItem* Teil einer solchen Fensterliste ist oder nicht.

Das Anordnen der Kindfenster implementieren

Zum Schluss soll natürlich auch verraten werden, wie sich die MDI-Fenster »in einem Aufwasch« zum Beispiel nebeneinander oder überlappend anordnen lassen. Das erledigt die *LayoutMdi*-Methode des Formulars, welcher der Modus als Parameter übergeben wird. Der folgende Aufruf ordnet die Fenster nebeneinander an:

```
Me.LayoutMdi(MdiLayout.TileHorizontal)
```

Abbildung 9.25:
Ein Bildbetrachter
als typische MDI-
Anwendung



9.9 Das fehlende Chart-Control

Abschließend muss leider auf eine kleine Lücke in dem ansonsten so umfangreichen Windows Forms-Modell hingewiesen werden. Aus welchen Gründen auch immer hat sich Microsoft dazu entschieden, kein Chart-Control mitzuliefern. Grundsätzlich ist das natürlich kein Problem, denn es gibt eine reichhaltige Auswahl an entsprechenden Controls. Das Dumme ist nur, dass die allermeisten nicht gerade preiswert sind (dafür aber auch einiges zu bieten haben). Doch es gibt eine Ausnahme: *Chart FX Lite* von Software FX. Laden Sie das Control über die URL <http://www.softwarefx.com/GetChartFXLite> herunter (Sie müssen sich dazu allerdings auf der Webseite registrieren und benötigen einen Schlüssel, der Ihnen per E-Mail zugesendet wird), und führen Sie die folgenden Schritte aus:

1. Legen Sie ein neues Projekt vom Typ WINDOWS-ANWENDUNG an.
2. Klicken Sie die Toolbox mit der rechten Maustaste an, und rufen Sie den Menüeintrag ELEMENT AUSWÄHLEN auf. Klicken Sie auf DURCHSUCHEN, und wählen Sie die Assembly *ChartFX.Lite.dll* im entsprechenden Verzeichnis aus (standardmäßig *C:\Programme\ChartFX Lite for .NET\bin*).





Abbildung 9.26:
Das Chart-Control
wird zur Toolbox
hinzugefügt.

3. Bestätigen Sie die Auswahl mit OK, und ordnen Sie das Chart-Control auf dem Formular an. Dabei ist zu beachten, dass das Control eine gewisse Mindestgröße besitzen muss. Ziehen Sie das Control entsprechend groß auf, ansonsten wird später nur ein leerer Rahmen oder lediglich ein Teil der Legende angezeigt.
4. Fügen Sie in die *Form_Load*-Prozedur die folgenden Befehle ein, die ein paar Zufallszahlen erzeugen:

```
Chart1.Gallery = SoftwareFX.ChartFX.Lite.Gallery.Pie
Chart1.Chart3D = True
Chart1.LegendBox = True
For i As Short = 1 To 10
    Chart1.Value(0, i) = New Random().Next(1, 100)
Next
```

Abbildung 9.27:
Das erfreut jeden
Programmierer –
ein kleines 3D-
Kuchendiagramm.



9.10 Zusammenfassung

Die Toolbox von Visual Basic 2005 bietet (fast) alles, was benötigt wird, um moderne Benutzeroberflächen mit relativ wenig Aufwand zu gestalten. Im Unterschied zu VB6 stehen diese Möglichkeiten allen .NET-Sprachen gleichermaßen zur Verfügung. Zu den beiden Dingen, die fehlen, gehören ein Chart-Control und eine »Guide Line«, also eine Art »Offizielles Handbuch zur Gestaltung moderner Windows-Oberflächen«⁹. Der erste Punkt lässt sich relativ schnell abhaken, da es Dutzende von Chart-Controls anderer Anbieter gibt (unter anderem das kostenlose Chart FX Lite), der zweite Punkt ist in erster Linie eine Sache der Erfahrung und eines gewissen Gespürs für Gestaltung, Farben und Ergonomie.

⁹ Das schreibe ich, glaube ich, schon seit über zehn Jahren. Irgendjemand sollte sich das einmal zu Herzen nehmen.